


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



# PAPER

# SOFT



sped. in abb. post. gr. 11/70

Anno I - N° 11 - 31 agosto 1984



**Pallavolo**  
**Dattilografia**



**Red Baron**  
**Pac-man**



**Battaglia navale**  
**Ritorno alla base**  
**Centipede**



**Mosaic puzzle**  
**Campo minato**



**Campo minato**  
**Il pescatore**

# Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numero seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	■
{G2}	2	■
{G3}	3	■
{G4}	4	■
{G5}	5	■
{G6}	6	■
{G7}	7	■
{G8}	8	■
{SG1}	CAPS SHIFT 1	■
{SG2}	CAPS SHIFT 2	■
{SG3}	CAPS SHIFT 3	■
{SG4}	CAPS SHIFT 4	■
{SG5}	CAPS SHIFT 5	■
{SG6}	CAPS SHIFT 6	■
{SG7}	CAPS SHIFT 7	■
{SG8}	CAPS SHIFT 8	■

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
A	A	Simbolo grafico definito nel programma in uso.
B	B	
C	C	
D	D	
E	E	
F	F	
G	G	
H	H	
I	I	
J	J	
K	K	
L	L	
M	M	
N	N	
O	O	
P	P	
Q	Q	
R	R	
S	S	
T	T	
U	U	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

# Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [ < > ] deve essere

premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME	■	{CYN}	CTRL 4	■	[<7>]	G 7	■
{HOME}	CLR/HOME	S	{PUR}	CTRL 5	■	[<8>]	G 8	■
{SU}	SHIFT ↑ CRSR ↓	■	{GRN}	CTRL 6	■	{F1}	f1	■
{GIU'}	↑ CRSR ↓	Q	{BLU}	CTRL 7	■	{F2}	f2	■
{SIN}	SHIFT ← CRSR →	I	{YEL}	CTRL 8	■	{F3}	f3	■
{DES}	← CRSR →	J	[<1>]	G 1	■	{F4}	f4	■
{RVS}	CTRL 9	R	[<2>]	G 2	■	{F5}	f5	■
{OFF}	CTRL 0	■	[<3>]	G 3	■	{F6}	f6	■
{BLK}	CTRL 1	■	[<4>]	G 4	■	{F7}	f7	■
{WHT}	CTRL 2	E	[<5>]	G 5	■	{F8}	f8	■
{RED}	CTRL 3	F	[<6>]	G 6	■			■



Paddle		<b>4</b>	<b>Pallavolo</b> di J. Higler trad. e adatt. di M. Cerofolini
		<b>7</b>	<b>Dattilografia</b> di A. Armstrong trad. e adatt. di M. Cerofolini
Extended Basic		<b>8</b>	<b>Red Baron</b> di W. K. Balthrop trad. e adatt. di
		<b>11</b>	<b>Puc-Man</b> di M. Bianchi
48K		<b>15</b>	<b>Battaglia navale</b> di P. C. Pasquali
16K/48K		<b>18</b>	<b>Ritorno alla base</b> di D. Naylor trad. e adatt. di C. Panzalis
16K/48K		<b>21</b>	<b>Centipede</b> di N. O'Brien trad. e adatt. di R. Quirico
		<b>22</b>	<b>Mosaic puzzle</b> di B. Jordan trad. e adatt. di M. Cristuib Grizzi
Joystick (opzionale per il C64)		<b>23</b>	<b>Campo minato</b> di S. Igo trad. e adatt. di F. Sarcina
3K ram per il VIC		<b>26</b>	<b>Campo minato</b> di S. Igo trad. e adatt. di F. Sarcina
		<b>29</b>	<b>Il pescatore</b> di D. Lacey trad. e adatt. di R. Comini

J. soft s.r.l.

**DIREZIONE, REDAZIONE,  
AMMINISTRAZIONE**

Via Rosellini, 12  
20124 MILANO  
Tel. (02) 68.88.228

**DIRETTORE RESPONSABILE:**

Pietro Dell'Orco

**COORDINAMENTO TECNICO:**

Riccardo Paolillo

**REDAZIONE:**

Lucio Bragagnolo  
Mauro Cristuib Grizzi

**GRAFICA E IMPAGINAZIONE:**

Margherita La Noce  
Raffaella Toffolatti

**FOTOCOMPOSIZIONE:**

d&b Via Vignola, 5  
Tel. 02/59.85.08  
20133 MILANO

**CONTABILITÀ:**

Giulia Pedrazzini  
Flavia Bonaiti

**AUTORIZZAZIONE ALLA  
PUBBLICAZIONE:**

Tribunale di Milano n° 200  
del 14.04.1984

**STAMPA:**

Elcograf  
Beverate (CO)

**PUBBLICITÀ**

Concessionario per l'Italia e l'Estero Reina s.r.l.  
Via Washington, 50  
20046 MILANO

Tel. (02) 49.88.066 (5 linee R.A.)  
Tlx. 316213 REINA I

Concessionario esclusivo per la  
DIFFUSIONE in Italia e Estero:  
SODIP - Via Zuretti, 25  
20125 MILANO

Spedizione in abbonamento  
postale Gruppo Il 70  
Prezzo della rivista L. 1.000  
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI  
RIPRODUZIONE O TRADUZIONE  
DEGLI ARTICOLI PUBBLICATI  
SONO RISERVATI

# Pallavolo

Questo è un gioco d'azione per due o quattro giocatori. Poiché difficilmente si hanno a disposizione più di due paddle, la maggior parte dei giochi è programmata per 1 o 2 concorrenti. Pallavolo invece riesce a superare questo ostacolo: due dei quattro giocatori utilizzeranno le paddle, mentre gli altri due interverranno nella partita per mezzo di tasti specifici. In questo modo si viene a creare una dicertente ed eccitante atmosfera di competizione tra squadre; nel caso non vi siano quattro giocatori disponibili il gioco risulterà forse anche più stimolante per i due o tre che vi prenderanno parte.

Dopo aver dato il RUN al programma, appariranno il campo da gioco ed i 4 giocatori. Il segno "" sotto un lato del campo indica la squadra che ha il servizio. I giocatori più vicini alla rete vengono chiamati schiacciatori mentre quelli più lontani palleggiatori. Questi ultimi sono controllati dalla manopola delle paddle. Lo schiacciatore a sinistra risponde ai tasti A e S mentre quello di destra è controllato dalle frecce che indicano la destra e la sinistra. I palleggiatori controllano il servizio

```

500 POKE 116,32: POKE 115,0
1000 GOSUB 2380
1010 GOSUB 2300
1020 GOTO 1550
1030 GOSUB 1100
1040 GOSUB 1150
1050 GOSUB 1240
1060 GOSUB 1290
1070 GOSUB 1380
1080 K = PEEK ( - 16384)
1090 POKE - 16368,0: GOTO 1030

1100 X1% = PDL (0) / 3
1110 IF X1% < 4 THEN X1% = 4
1120 IF X1% = XA% THEN 1140
1130 HCOLOR= 0: DRAW 2 AT XA%,1
12: HCOLOR= 3: DRAW 2 AT X1%,112
:XA% = X1%
1140 RETURN
1150 IF SB% = 0 THEN 1200

```

che parte quando si preme il bottone della paddle. Le vostre possibilità di controllare il movimento dei giocatori sono maggiori di quanto non immaginate: per esempio premendo il tasto A una volta lo schiacciatore di sinistra si sposta verso sinistra, premendo di nuovo lo stesso tasto esso si muoverà nella medesima direzione, ma molto più velocemente. Se poi volete rallentare o fermare il suo movimento basterà premere il tasto S rispettivamente 1 o 2 volte.

Si seguono le comuni regole della pallavolo. Bisogna totalizzare 21 punti e la squadra vincitrice deve avere almeno 2 punti di distacco. Si segna un punto solo quando si è avuto il servizio. Se una squadra dopo tre tentativi non è riuscita a lanciare la palla al di là della rete perde il punto. Attenzione: contrariamente alle regole convenzionali lo stesso giocatore può colpire la palla più di una volta di seguito; infatti in ogni squadra vi sono due giocatori i quali, inoltre non possono lasciare le rispettive zone d'azione.

Buon divertimento e vinca il migliore!

```

1160 X1% = XB% + SB%
1170 IF X1% < 94 THEN X1% = 94:
SB% = 0
1180 IF X1% > 134 THEN X1% = 13
4:SB% = 0
1190 HCOLOR= 0: DRAW 3 AT XB%,1
12: HCOLOR= 3: DRAW 3 AT X1%,112
:XB% = X1%
1200 IF K < > 193 AND K < > 2
11 THEN 1230
1210 IF K = 193 THEN SB% = SB%
- 4: GOTO 1230
1220 SB% = SB% + 4
1230 RETURN
1240 X1% = PDL (1) / 3 + 190
1250 IF X1% < 194 THEN X1% = 19
4
1260 IF X1% = XD% THEN 1280
1270 HCOLOR= 0: DRAW 4 AT XD%,1
12: HCOLOR= 3: DRAW 4 AT X1%,112

```

```

:XD% = X1%
1280 RETURN
1290 IF SD% = 0 THEN 1340
1300 X1% = XC% + SD%
1310 IF X1% < 145 THEN X1% = 145:SD% = 0
1320 IF X1% > 185 THEN X1% = 185:SD% = 0
1330 HCOLOR= 0: DRAW 5 AT XC%,112: HCOLOR= 3: DRAW 5 AT X1%,112:XC% = X1%
1340 IF K < 136 AND K > 149 THEN 1370
1350 IF K = 136 THEN SD% = SD% - 4: GOTO 1370
1360 SD% = SD% + 4
1370 RETURN
1380 B1% = BX% + SX%:B2% = BY% + SY%
1390 B% = 8 * RND (1):C% = SGN (.5 - RND (1)) * B%
1400 SY% = SY% + 4
1410 IF ABS (B1% - 139) < 5 AND B2% > 90 THEN B1% = BX% - SX%:SX% = SX%: GOSUB 2240
1420 IF B2% > 105 AND BY% < 106 THEN 1460
1430 IF B2% > 105 THEN 2020
1440 IF B1% < 4 OR B1% > 275 THEN 2020
EN 2020
1450 GOTO 1508
1460 IF ABS (B1% - XA%) < 6 OR ABS (B1% - XB%) < 6 THEN B1% = B1% + 4: SX% = 4 + B%: SY% = A%: GOSUB 2260: B2% = 105: H = 1: GOTO 1500
1470 IF ABS (B1% - XA%) < 10 OR ABS (B1% - XB%) < 10 THEN B1% = B1% + 2: SX% = 2 + C%: SY% = 2 * A% / 3: GOSUB 2260: B2% = 105: H = 1: GOTO 1500
1480 IF ABS (B1% - XC%) < 6 OR ABS (B1% - XD%) < 6 THEN B1% = B1% - 4: SX% = - 4 - B%: SY% = A%: GOSUB 2260: B2% = 105: H = 2: GOTO 1500
1490 IF ABS (B1% - XC%) < 10 OR ABS (B1% - SD%) < 10 THEN B1% = B1% - 2: SX% = - 2 - C%: SY% = 2 * A% / 3: GOSUB 2260: B2% = 105: H = 2: GOTO 1500
1495 GOTO 1508
1500 IF LH = H THEN NH = NH + 1: IF NH > 3 THEN H3 = H: GOTO 2020
1505 IF LH > H THEN LH = H: NH = 1
1507 VTAB 21: HTAB 21: PRINT NH
1508 IF B2% > 105 THEN B2% = 115
1510 IF B1% < 4 OR B1% > 275 THEN 2020
1520 HCOLOR= 0: DRAW 1 AT BX%,BY%: HCOLOR= 3: DRAW 1 AT B1%,B2%:BX% = B1%:BY% = B2%
1530 RETURN
1550 A% = - 24
1560 HGR
1570 HOME
1580 VTAB 22
1590 HCOLOR= 1
1600 HPLOT 0,120 TO 279,120
1610 HCOLOR= 5: HPLOT 0,121 TO 279,121
1620 HCOLOR= 1: HPLOT 0,122 TO 279,122
1630 HCOLOR= 1: HPLOT 0,123 TO 279,123
1640 HCOLOR= 3
1650 ROT= 0: SCALE= 1
1660 HPLOT 140,90 TO 140,118
1670 HPLOT 139,90 TO 139,118
1680 GOSUB 2400
1690 VTAB 22: CALL - 868
1700 POKE 232,0: POKE 233,27
1710 XA% = PDL (0) / 3
1720 IF XA% < 4 THEN XA% = 4
1730 DRAW 2 AT XA%,112
1740 XB% = 110
1750 DRAW 3 AT XB%,112
1760 XD% = PDL (1) / 3 + 190
1770 IF XD% < 194 THEN XD% = 194
1780 DRAW 4 AT XD%,112
1790 XC% = 169
1800 DRAW 5 AT XC%,112
1810 S1 = 0: S2 = 0
1820 VTAB 22: HTAB 10: PRINT S1: VTAB 22: HTAB 30: PRINT S2
1830 SB% = 0: SD% = 0
1840 K = 0
1850 S = 1
1860 R = RND (1): IF R < .5 THEN S = 2
1870 GOSUB 1890
1880 GOTO 1030
1890 B% = 8 * RND (1)
1900 IF S = 1 THEN BX% = 85: BY% = 100: SX% = 5 + B%: SY% = A%: VTAB 21: HTAB 2: PRINT "^": VTAB 21: HTAB 38: PRINT " "
1910 IF S = 2 THEN BX% = 194: BY% = 100: SX% = - (5 + B%): SY% = A%: VTAB 21: HTAB 2: PRINT " ": VTAB 21: HTAB 38: PRINT "^"
1920 H = S
1930 IF S = 1 THEN I = PEEK (

```

```

- 16287): IF I < 128 THEN 1930
1940 IF S = 2 THEN I = PEEK (
- 16286): IF I < 128 THEN 1940
1950 GOSUB 2240
1960 SB% = 0:SD% = 0
1965 H3 = 0:LH = 0:NH = 0
1967 VTAB 21: HTAB 21: PRINT "
"
1970 K = 0: POKE - 16368,0
1980 HCOLOR= 3
1990 HPLOT 140,90 TO 140,118
2000 HPLOT 139,90 TO 139,118
2010 RETURN
2020 IF H3 = 1 THEN 2080
2022 IF H3 = 2 THEN 2040
2025 IF BX% < 139 THEN 2070
2030 IF H = 1 AND B1% > 275 THE
N 2080
2040 IF S = 1 THEN S1 = S1 + 1
2050 S = 1
2060 GOTO 2100
2070 IF H = 2 AND B1% < 4 THEN
2040
2080 IF S = 2 THEN S2 = S2 + 1
2090 S = 2
2100 FOR W = 1 TO 20: GOSUB 224
0: NEXT
2110 VTAB 22: HTAB 10: PRINT S1
: VTAB 22: HTAB 30: PRINT S2
2120 IF (S1 > 14 OR S2 > 14) AN
D ABS (S1 - S2) > 1 THEN 2150
2130 HCOLOR= 0: DRAW 1 AT BX%,B
Y%: HCOLOR= 3
- 2140 POP : GOSUB 1890: GOTO 103
0
2150 VTAB 21: HTAB 19: FLASH :
PRINT "GAME"
2160 HTAB 22: HTAB 19: PRINT "O
VER"
2170 NORMAL
2180 HCOLOR= 2: HPLOT 10,10 TO
20,20: CALL 62454: HCOLOR= 0: FO
R I = 120 TO 123: HPLOT 0,I TO 2
79,I: NEXT : HPLOT 139,90 TO 139
,118: HPLOT 140,90 TO 140,118
2190 FOR I = 150 TO 159: HPLOT
0,I TO 279,I: NEXT
2200 FOR I = 1 TO 3: CALL - 19
8: NEXT
2210 K = PEEK ( - 16384): IF K
< 128 THEN 2210
2220 POKE - 16368,0: IF K < >
141 THEN 2210
2230 GOTO 1550
2240 ZZ = PEEK ( - 16336) + PE
EK ( - 16336) - PEEK ( - 16336)
2250 RETURN
2260 N% = B1%: IF N% > 255 THEN

```

```

N% = 255
2270 IF N% < 4 THEN N% = 4
2280 POKE 776,N%: POKE 777,10:
CALL 778
2290 RETURN
2300 TEXT : HOME : VTAB 10: HTA
B 10: PRINT "P A L L A V O L O"
2310 HTAB 12: VTAB 12: PRINT "D
I JIM HILGER"
2320 VTAB 20: PRINT "LE PADDLES
MUOVONO GLI ESTERNI ": PRINT "I
TASTI A S E -> <- MUOVONO GLI I
NTERNI"
2330 PRINT "IL LATO CON LA FREC
CIA ^ ": PRINT "SERVE PREMENDO I
L PULSANTE. RETURN PER INIZIARE
"
2340 FOR I = 10 TO I STEP - 1:
B1% = I * 10: GOSUB 2260: NEXT :
FOR I = 1 TO 50: NEXT : FOR I =
1 TO 10:B1% = I * 10: GOSUB 226
0: NEXT
2350 K = PEEK ( - 16384): IF K
< 128 THEN 2350
2360 POKE - 16368,0: IF K < >
141 THEN 2350
2370 RETURN
2380 POKE 776,255: POKE 777,255
: POKE 778,173: POKE 779,48: POK
E 780,192: POKE 781,136: POKE 78
2,208: POKE 783,5: POKE 784,206:
POKE 785,9: POKE 786,3: POKE 78
7,240
2390 POKE 788,9: POKE 789,202:
POKE 790,208: POKE 791,245: POKE
792,174: POKE 793,8: POKE 794,3
: POKE 795,76: POKE 796,10: POKE
797,3: POKE 798,96: RETURN
2400 FOR I = 1 TO 284: READ I%:
POKE 6912 + I - 1,I%: NEXT I: R
ESTORE : RETURN
2410 DATA 5,0,12,0,42,0,98,0,
163,0,219,0,27,27,18,10,41,45,13
,24
2420 DATA 31,27,59,40,9,9,9,60
,27,27,27,12,13,9,41,24,59,63,31
,0
2430 DATA 0,24,27,27,18,18,18,
45,13,45,5,24,27,31,59,8,13,45,9
,24
2440 DATA 27,59,63,8,45,13,9,2
4,27,63,63,44,45,45,9,24,59,63,6
3,8
2450 DATA 41,9,45,24,31,59,63,
8,45,13,41,24,59,27,27,0,0,0,27,
27
2460 DATA 18,18,18,45,13,9,45,
28,31,27,27,7,8,41,13,45,1,24,27
,63

```

2470 DATA 31,3,8,41,45,45,1,24  
 ,59,63,63,3,8,41,45,45,1,24,31,5  
 9  
 2480 DATA 27,7,40,13,45,13,37,  
 31,59,63,27,44,9,9,9,33,63,31,27  
 ,63  
 2490 DATA 4,0,0,27,27,18,18,18  
 ,41,45,41,37,59,27,31,3,8,9,45,4  
 1  
 2500 DATA 24,63,31,27,8,9,41,4  
 5,56,63,63,27,8,9,45,45,28,63,63  
 ,31  
 2510 DATA 8,45,9,13,24,63,31,5

9,8,13,41,45,24,27,27,31,0,0,0,2  
 7  
 2520 DATA 27,18,18,18,45,13,9,  
 45,28,31,27,27,7,8,41,13,45,1,24  
 ,27  
 2530 DATA 63,31,3,8,41,45,45,1  
 ,24,59,63,63,3,8,41,45,45,1,24,3  
 1  
 2540 DATA 59,27,7,40,13,45,13,  
 37,31,59,63,27,44,9,9,9,33,63,31  
 ,27  
 2550 DATA 63,4,0,0,3,0,0,0,194  
 ,216,0,194,0,0,0,194,217,0,100,0



## Dattilografia

Imparare a battere a macchina velocemente è certamente molto utile: con questo programma riuscirete ad acquistare l'automatismo necessario per diventare rapidissimi dattilografi solo con un po' di esercizio, senza alcuna difficoltà. In futuro eviterete di perdere tempo a cercare i caratteri sulla tastiera. È importante che usiate sempre le stesse dita per gli stessi tasti così che il processo di apprendimento diventi meccanico. Dapprima il computer vi chiederà quali lettere o caratteri volete esercitare: in questo modo potrete imparare un po' per volta ed evitare di affrontare fin dall'inizio l'intera tastiera. Ai principianti generalmente si consiglia di iniziare con le lettere ASDFG ed in un secondo momento con HJKL per poi esercitare entrambi i gruppi contemporaneamente. ASDFGHJKL. Comunque tenete sempre i polpastrelli delle dita che non usate su queste lettere: saranno utili punti di riferimento per trovare gli altri tasti. Continuando con QWASZX oppure con TYGHBN acquisirete una certa pratica nel trovare le lettere in alto ed in basso nella tastiera: è un errore imparare ogni fila separatamente perché perdereste la percezione della distanza dei tasti disposti su file diverse.

Il computer vi chiederà "Quante lettere?" e voi dovete stabilire quanti caratteri esercita-

re allo stesso tempo. È consigliabile aumentare gradatamente la lunghezza delle parole senza preoccuparsi troppo della velocità che risulterà naturale conseguenza dell'automatismo da voi acquistato. Il computer vi chiederà anche se volete l'eco cioè se desiderate vedere sul monitor le lettere che battere sulla tastiera. Dovrete comunque esercitarvi senza guardare mai la tastiera, leggendo solo il testo che state copiando. Al fine di un rapido recupero dell'errore, ogni volta che sbagliate il programma vi segnala l'errore; inoltre la parola successiva conterrà il carattere che dovrete esercitare ulteriormente. Infine il computer vi chiederà quante volte desiderate esercitare la stessa parola prima che vi venga fornito il vostro punteggio. Al termine, oltre al punteggio che avrete meritato, vi verrà mostrato anche l'elenco degli errori che avete commesso.

Posizioni consigliate:

1	2	3	4	5	6	7	8	9	0
Q	W	E	R	T	Y	U	I	O	P
A	S	D	F	G	H	J	K	L	;
Z	X	C	V	B	N	M	,	.	/

mign anu med indice indice med an-  
 m i g n

SINISTRA

DESTRA

Con i due pollici la barra spaziatrice.

```

10 REM ESERCIZIO DI SCRITTURA
50 TEXT : HOME
100 INPUT "QUALI CARATTERI VUOI
? ";C$
105 L = LEN (C$): DIM A(L)
110 INPUT "LUNGHEZZA DELLE PARO
LE? ";WL
120 INPUT "VUOI L'ECO? ";A$
125 IF LEFT$ (A$,1) = "S" THEN
E = 1
130 INPUT "QUANTI TENTATIVI? ";
NT
200 FOR T = 1 TO NT
210 NP = NP + WL
220 A$ = ""
230 FOR I = 1 TO WL
240 P = INT (L * RND (1) + 1)
250 A$ = A$ + MID$ (C$,P,1)
260 NEXT
270 PRINT : PRINT A$
300 FOR I = 1 TO WL
310 GET B$
320 IF E = 0 THEN 350
330 PRINT B$;
350 IF B$ < > MID$ (A$,I,1) T
HEN 500
360 NR = NR + 1

```

```

370 NEXT
375 IF E = 1 THEN PRINT
380 NEXT T
400 PRINT : PRINT "IL TUO PUNTE
GGIO E' "; INT (100 * NR / NP);"
%"
402 IF NR = NP THEN 415
405 PRINT "ERRORI: ": FOR I = 1
TO L: IF A(I) = 0 THEN 410
407 PRINT MID$ (C$,I,1);" - ";
A(I);" VOLTE"
410 NEXT
415 PRINT : INPUT "VUOI CONTINU
ARE? ";A$
420 IF LEFT$ (A$,1) = "S" THEN
RUN
430 END
500 FOR J = 1 TO L
510 IF MID$ (C$,J,1) < > MID
$ (A$,I,1) THEN NEXT : GOTO 520
515 A(J) = A(J) + 1
520 PRINT : PRINT "ERRORE SU ";
MID$ (A$,I,1)
530 FOR I = 1 TO 300: NEXT
540 GOTO 380

```

## Red Baron



Extended Basic

*Chi non ha letto almeno una volta nella sua vita una strip di Snoopy in "sella" alla propria cuccia nell'emulazione dei vecchi duelli aerei tra biplani?*

*Bene, ora avete a disposizione un ottimo programma in Extended Basic per due persone, che inserito nel vostro TI-99 vi darà modo di disporre di un vecchio ma sempre efficiente Sopwith Camel con il quale potrete duellare nello spazio libero contro il temibile Barone Rosso su quattro differenti livelli di gioco e con la possibilità di scegliere un duello notturno o diurno.*

*L'unico handicap di cui verrete caricati è la mancanza di installazione del mitragliatore di bordo, per cui, per abbattere il nemico non avrete altra scelta che impugnare la vo-*

*stra pistola e sparare all'avversario, con tutti i vantaggi/svantaggi dell'uso di una tale arma.*

*Addirittura, usando la quarta opzione, correrete il rischio di venire abbattuti dalle pallottole vaganti dal vostro avversario diverso tempo prima.*

*Per effettuare i movimenti del giocatore n° 1, servirsi dei tasti con le frecce (E-S-X-D), mentre per quelli del n° 2 usare i tasti I-J-M-K; per sparare il numero 1 userà il tasto F, mentre il secondo giocatore dovrà servirsi del tasto H.*

*Detto questo non vi rimane che battere il listato qui di seguito e salire sul vostro biplano. Buon combattimento!*



```

100 REM *****
110 REM * RED BARON *
120 REM *****
130 REM
140 REM
150 REM
160 REM
170 CALL CLEAR :: CALL SCREEN(6)

180 P=1 :: P1=5
190 CALL CHAR(96,"0000000081C0E2
F73F3E07102000000000000000F8017D
FDF05FD217070")
200 CALL CHAR(100,"0000010204090
10B0706FFFF7E4A02022090086CFAE9D
8B07CDC98")
210 CALL CHAR(104,"0F01077FFFFFFF
FFF7F06070703060E1EE000C0FCFEFEF
EFEFCC0C0C080C0E0F0")
220 CALL CHAR(108,"040910365F971
B0D3E3B19000000000000000804020908
0D0E060FFFF7C524040")
230 CALL CHAR(112,"000000001F80B
EBFFFA0BF840E0E000000000000081034
7EFCFCF00804")
240 CALL CHAR(116,"0000000005081
1274F1E9D4B3C100804303C383CF8FFB
060C080E0E0C0")
250 CALL CHAR(120,"0F07030103030
33F7F7F7F7F3F030007787060C0E0E06
0FEFFFFFFFEE080F0")
260 CALL CHAR(124,"0C3C1C1C3C1FF
F0D0603010707030000000000000A0108
8E4F278B9D27C081020")
270 CALL CHAR(128,"00000000000000
00101000000000000000000000000000
08080")
280 CALL CHAR(132,"FC7E3C38180C0
4012030301C1F030000FC783870F0604
080040C0C38F0C0")
290 CALL MAGNIFY(3)
300 DISPLAY AT(12,9):"RED BARON"

310 CALL SPRITE(#1,96,2,120,20,0
,5,#2,112,2,70,240,0,-5)
320 FOR TD=1 TO 2 :: CALL SOUND(
1500,110,2,220,2,1000,30,-8,2)
330 NEXT TD :: CALL SOUND(1,9999
,30)
340 CALL DELSPRITE(#1,#2):: CALL
CLEAR
350 DISPLAY AT(5,1):"NOME PRIMO
GIOCATORE?" :: ACCEPT AT(7,3)S
IZE(8):PLA1$
360 DISPLAY AT(11,1):"NOME SECON
DO GIOCATORE?" :: ACCEPT AT(13,3
)SIZE(8):PLA2$
370 CALL CLEAR
380 DISPLAY AT(4,3):"LIVELLI DI
GIOCO:"

```

```

390 DISPLAY AT(6,5):"1. LATTANTE
" :: DISPLAY AT(8,5):"2. ESPERTO
"
400 DISPLAY AT(10,5):"3. VECCHIA
VOLPE" :: DISPLAY AT(12,5):"4.
VS. RED BARON"
410 CALL KEY(0,KEY,S)
420 IF KEY<49 OR KEY>52 THEN 410
ELSE V=2*(KEY-48)
430 CALL CLEAR :: DISPLAY AT(5,3
):"PREFERISCI UNA MISSIONE D
IURNA O NOTTURNA? (D/N)"
440 CALL KEY(0,KEY,S)
450 IF KEY=68 THEN CO1=6 :: CO2=
2 :: GOTO 470
460 IF KEY=78 THEN CO1=2 :: CO2=
6 ELSE 440
470 CALL CLEAR
480 CALL SCREEN(CO1)
490 CALL SPRITE(#1,96,8,70,16,0,
V,#2,112,11,70,240,0,-V)
500 CALL KEY(1,K1,S1):: CALL KEY
(2,K2,S2)
510 IF S1=0 AND S2=0 THEN 820
520 IF S1=0 THEN 680
530 IF K1<>5 THEN 570 ELSE IF P>
3 AND P<8 THEN P=P-1 ELSE IF P<3
OR P=8 THEN P=P+1
540 IF P=0 THEN P=8
550 IF P=9 THEN P=1
560 GOTO 650
570 IF K1<>0 THEN 610 ELSE IF P>
3 AND P<7 THEN P=P+1 ELSE IF P<3
OR P=8 THEN P=P-1
580 IF P=9 THEN P=1
590 IF P=0 THEN P=8
600 GOTO 650
610 IF K1<>2 THEN 630 ELSE IF P>
5 THEN P=P-1 ELSE IF P<5 AND P>1
THEN P=P+1
620 GOTO 650
630 IF K1<>3 THEN 660 ELSE IF P<
5 AND P>1 THEN P=P-1 ELSE IF P>5
THEN P=P+1
640 IF P=9 THEN P=1
650 CALL PATTERN(#1,(P*4)+92)::
ON P GOSUB 910,920,930,940,950,9
60,970,980 :: GOTO 670
660 IF K1=12 THEN GOSUB 990
670 IF S2=0 THEN 820
680 IF K2<>5 THEN 720 ELSE IF P1
>3 AND P1<8 THEN P1=P1-1 ELSE IF
P1<3 OR P1=8 THEN P1=P1+1
690 IF P1=0 THEN P1=8
700 IF P1=9 THEN P1=1
710 GOTO 800
720 IF K2<>0 THEN 760 ELSE IF P1
>3 AND P1<7 THEN P1=P1+1 ELSE IF
P1<3 OR P1=8 THEN P1=P1-1
730 IF P1=0 THEN P1=8

```

```

740 IF P1=9 THEN P1=1
750 GOTO 800
760 IF K2<>2 THEN 780 ELSE IF P1
>5 THEN P1=P1-1 ELSE IF P1<5 AND
P1>1 THEN P1=P1+1
770 GOTO 800
780 IF K2<>3 THEN 810 ELSE IF P1
<5 AND P1>1 THEN P1=P1-1 ELSE IF
P1>1 THEN P1=P1+1
790 IF P1=9 THEN P1=1
800 CALL PATTERN(#2,(P1*4)+92)::
ON P1 GOSUB 1110,1120,1130,1140
,1150,1160,1170,
1180 :: GOTO 820
810 IF K2=1 THEN GOSUB 1190
820 CALL COINC(#2,#3,3*V,PC):: I
F PC=-1 THEN 1500
830 CALL COINC(#1,#4,3*V,PC):: I
F PC=-1 THEN 1350
840 CALL COINC(#1,#2,2.5*V,PC)::
IF PC=-1 THEN 1220
850 CALL SOUND(-4250,110,2,220,2
,-8,2)
860 IF SH1=0 THEN 880
870 CALL POSITION(#3,B3,B4):: IF
B3<6 OR B3>186 OR B4<8 OR B4>25
0 THEN CALL DELSPRITE(#3):: SH1=
0
880 IF SH2=0 THEN 900
890 CALL POSITION(#4,C3,C4):: IF
C3<6 OR C3>186 OR C4<8 OR C4>25
0 THEN CALL DELSPRITE(#4):: SH2=
0
900 GOTO 500
910 CALL MOTION(#1,0,V):: RETURN

920 CALL MOTION(#1,-V*.6,V*.6)::
RETURN
930 CALL MOTION(#1,-V,0):: RETUR
N
940 CALL MOTION(#1,-V*.6,-V*.6):
: RETURN
950 CALL MOTION(#1,0,-V):: RETUR
N
960 CALL MOTION(#1,V*.6,-V*.6)::
RETURN
970 CALL MOTION(#1,V,0):: RETURN

980 CALL MOTION(#1,V*.6,V*.6)::
RETURN
990 IF SH1=1 THEN RETURN
1000 CALL POSITION(#1,B1,B2)
1010 P3=P :: SP=3 :: A1=B1 :: A2
=B2 :: SH1=1
1020 ON P3 GOTO 1030,1040,1050,1
060,1070,1080,1090,1100
1030 CALL SPRITE(#SP,128,CO2,A1,
A2,0,V*2):: RETURN
1040 CALL SPRITE(#SP,128,CO2,A1,
A2,-V*1.6,V*1.6):: RETURN

```

```

1050 CALL SPRITE(#SP,128,CO2,A1,
A2,-V*2,0):: RETURN

1060 CALL SPRITE(#SP,128,CO2,A1,
A2,-V*1.6,-V*1.6):: RETURN
1070 CALL SPRITE(#SP,128,CO2,A1,
A2,0,-V*2):: RETURN
1080 CALL SPRITE(#SP,128,CO2,A1,
A2,V*1.6,-V*1.6):: RETURN
1090 CALL SPRITE(#SP,128,CO2,A1,
A2,V*2,0):: RETURN
1100 CALL SPRITE(#SP,128,CO2,A1,
A2,V*1.6,V*1.6):: RETURN
1110 CALL MOTION(#2,0,V):: RETUR
N
1120 CALL MOTION(#2,-V*.6,V*.6):
: RETURN
1130 CALL MOTION(#2,-V,0):: RETU
RN
1140 CALL MOTION(#2,-V*.6,-V*.6)
:: RETURN
1150 CALL MOTION(#2,0,-V):: RETU
RN
1160 CALL MOTION(#2,V*.6,-V*.6):
: RETURN
1170 CALL MOTION(#2,V,0):: RETUR
N
1180 CALL MOTION(#2,V*.6,V*.6)::
RETURN
1190 IF SH2=1 THEN RETURN
1200 CALL POSITION(#2,C1,C2)
1210 A1=C1 :: A2=C2 :: P3=P1 ::
SP=4 :: SH2=1 :: GOTO 1020
1220 CALL MOTION(#1,0,0,#2,0,0):
: CALL POSITION(#1,PO1,PO2,#2,PO
3,PO4)
1230 CALL SPRITE(#1,132,10,PO1,P
O2,20,0,#2,132,10,PO3,PO4,20,0)

1240 CALL POSITION(#1,CR1,CR2)
1250 CALL POSITION(#2,CR3,CR4)
1260 IF CR1>186 OR CR3>186 THEN
CALL DELSPRITE(#1,#2)
1270 CALL SOUND(-1000,110,2,220,
2,-7,0)
1280 IF CR1<>0 OR CR3<>0 THEN 12
40
1290 CALL DELSPRITE(#1,#2,#3,#4)

1300 CALL SCREEN(6)
1310 DISPLAY AT(5,13):"CRAAAASSS
H"
1320 DISPLAY AT(10,5):"ENTRAMBI
IN COLLISIONE!"
1330 DISPLAY AT(12,9):"NESSUN PU
NTEGGIO TOTALIZZATO"
1340 GOTO 1640
1350 CALL DELSPRITE(#3,#4)
1360 CALL MOTION(#1,0,0):: CALL
POSITION(#1,PO1,PO2)

```

```

1370 CALL SPRITE(#1,132,10,PO1,P
02,20,0)
1380 CALL SOUND(-1000,440,2,220,
2,-4,0)
1390 CALL POSITION(#1,CR1,CR2)
1400 IF CR1<186 THEN 1380
1410 CALL DELSPRITE(#1,#2)
1420 RED=RED+1
1430 CALL SCREEN(6)
1440 DISPLAY AT(5,1):PLA2$;" HA
VINTO LA BATTAGLIA."
1450 DISPLAY AT(9,3):PLA2$;" PUN
TI : "
1460 DISPLAY AT(9,26):RED
1470 DISPLAY AT(11,3):PLA1$;" PU
NTI : "
1480 DISPLAY AT(11,26):BLUE
1490 GOTO 1640
1500 CALL DELSPRITE(#3,#4)
1510 CALL MOTION(#2,0,0):: CALL
POSITION(#2,PO1,PO2)
1520 CALL SPRITE(#2,132,10,PO1,P
02,20,0)
1530 CALL SOUND(-1000,440,2,220,
2,-4,0)
1540 CALL POSITION(#2,CR3,CR4)
1550 IF CR3<186 THEN 1530
1560 CALL DELSPRITE(#2,#1)
1570 BLUE=BLUE+1

```

```

1580 CALL SCREEN(6)
1590 DISPLAY AT(5,1):PLA1$;" HA
VINTO LA BATTAGLIA."
1600 DISPLAY AT(8,3):PLA1$;" PUN
TI : "
1610 DISPLAY AT(8,26):BLUE
1620 DISPLAY AT(10,3):PLA2$;" PU
NTI : "
1630 DISPLAY AT(10,26):RED
1640 IF RED=10 OR BLUE=10 THEN 1
690
1650 DISPLAY AT(20,3):"PRONTO PE
R RIPARTIRE? (S/N)"
1660 PC=0 :: SH1=0 :: SH2=0 :: P
=1 :: P1=5
1670 CALL KEY(0,KEY,S)
1680 IF KEY=83 THEN 370 ELSE IF
KEY<>78 THEN 1670
1690 IF BLUE>RED THEN PRINT PLA1
$;" HA VINTO LA GUERRA.":"SPIACE
NTE, ";PLA2$ :: GOTO 1720
1700 IF RED>BLUE THEN PRINT PLA2
$;" HA VINTO LA GUERRA.":"SPIACE
NTE, ";PLA1$ :: GOTO 1720
1710 PRINT "LA GUERRA SI E'CONCL
USA          SENZA VINCITORI.
          ENTRAMBE LE SQUADRE SONO ORA
IN PACE."
1720 END

```

## Puc-Man



*Puc-Man è sicuramente il "padre" di quei video-giochi di inseguimento e di strategia, che si svolgono in un labirinto.*

*È stato per un lungo periodo di tempo il più appassionante dei video-games. Infatti una buona parte dei computer-games, che si trovano oggi nei bar e nelle sale-giochi (senza contare quelli creati dalle varie software-house) non sono altro che una variante più elaborata e sempre più completa del "vecchio" puc-man.*

*Dato che il suddetto gioco ancor oggi piace, ho deciso di farne una versione per l'home computer TI-99/4A ed il programma è risultato discreto, soprattutto per quanto riguarda le routines "intelligenti" di inseguimento.*

*Dato il run, viene chiesto da programma il numero dei fantasmi contro i quali il nostro puc-man dovrà destreggiarsi (il numero varia da uno a quattro). Dopo di che, viene richiesto il livello di difficoltà, che è normale o difficile; la differenza fra i due livelli è la seguente: inserendo il livello normale, i fantasmi attraversano i tunnel del dedalo alla stessa velocità del puc-man, mentre inserendo il livello difficile, i fantasmi attraversano talvolta le pareti del labirinto, infastidendo non poco il mangiatore di puntini.*

*Dopo questa prima fase, vi è la stampa del labirinto con i relativi puntini e lampeggiatori; infine ha inizio il gioco.*

*In un primo tempo, i fantasmi si muovono*

casualmente nel labirinto; dopo un po', entra in azione la routine di inseguimento, che li fa convergere verso il puc-man al quale non restano che due scelte: scappare o ricaricarsi

```

100 REM *****
110 REM * *
120 REM * PUC-MAN GAME *
130 REM * *
140 REM *****
150 REM per TI-99/4A
      ti-basic      memori
a richiesta [15 KB]
160 REM copyright:
      MASSIMO BIANCHI
      tel 031/701363
170 CALL CLEAR
180 CALL SCREEN(8)
190 CALL COLOR(1,4,1)
200 CALL COLOR(2,15,1)
210 CALL COLOR(12,16,1)
220 CALL COLOR(10,16,1)
230 FOR H=3 TO 8
240 CALL COLOR(H,16,1)
250 NEXT H
260 CALL COLOR(13,14,1)
270 CALL CHAR(128,"7E5ADBFFD5EBF
F55")
280 N_PIC=3
290 CP=0
300 PUN=0
310 RANDOMIZE
320 CASO=0
330 CALL CHAR(124,"3C7EF8E0E0F87
E3C")
340 PRINT "P U C -- M A N -- G A
M E:":: " | ";CHR$(128)
;" ";CHR$(128);" ";CHR$(128);" "
;CHR$(128)::
350 INPUT "INSERISCI IL NUMERO D
EI FAN-TASMINI 1-4 ":NFAN
360 IF (NFAN<>1)*(NFAN<>2)*(NFAN
<>3)*(NFAN<>4) THEN 350
370 PRINT ::"INSERISCI IL LIVELL
O GIOCO:"::
380 INPUT "1 NORMALE

      2 DIFFICILE ...":LDIFF
390 IF (LDIFF<>1)*(LDIFF<>2) THEN
370
400 LDIFF=LDIFF-2
410 CALL CLEAR
420 CALL CHAR(43,"0000001818")
430 CALL CHAR(104,"00003C3C3C3C"
)
440 CALL CHAR(42,"55AA55AA55AA55
AA")
450 CALL CHAR(96,"003C7E7E7E7E3C
")

```

con uno dei quattro stimolatori. Dopo di che... per sapere il resto, caricate il programma e... buon gioco!

```

460 BB=99
470 PCP=1
480 DIM M(20,24),PC(100),BIN(100
),PIC(32),CA(4),CB(4),CAV(4),CBV
(4)
490 FOR H=1 TO 100
500 PC(H)=INT(RND*4)+1
510 NEXT H
520 CALL COLOR(10,14,1)
530 FOR H=1 TO 100
540 BIN(H)=INT(RND*2)+1
550 NEXT H
560 PRINT " PUC-MAN PUNTI:00
0"::
570 PRINT " *****
****"
580 PRINT " *
*"
590 FOR H=1 TO 4
600 PRINT " * *** *** *** *
** *"
610 PRINT " * *** *** *
** *"
620 PRINT " * *** *** *** *
** *"
630 PRINT " * *** ***
*"
640 NEXT H
650 PRINT " *****
****"
660 CALL HCHAR(22,11,32,11)
670 REM ** STAMPA DEI LAMPEGGIA
TORI **
680 CALL HCHAR(7,12,32)
690 CALL HCHAR(8,12,96)
700 M(4,8)=96
710 CALL HCHAR(7,20,32)
720 CALL HCHAR(8,20,96)
730 M(4,16)=96
740 CALL HCHAR(21,12,32)
750 CALL HCHAR(20,12,96)
760 M(16,8)=96
770 CALL HCHAR(21,20,32)
780 CALL HCHAR(20,20,96)
790 M(16,16)=96
800 REM puntini
810 FOR H=6 TO 26 STEP 2
820 CALL HCHAR(6,H,104)
830 M(2,H-4)=104
840 CALL HCHAR(22,H,104)
850 M(18,H-4)=104
860 NEXT H
870 FOR H=6 TO 22 STEP 2
880 FOR X=6 TO 26 STEP 4

```

```

890 CALL HCHAR(H,X,104)
900 M(H-4,X-4)=104
910 NEXT X
920 NEXT H
930 FOR H=10 TO 18 STEP 4
940 FOR X=8 TO 24 STEP 8
950 CALL HCHAR(H,X,104)
960 M(H-4,X-4)=104
970 NEXT X
980 NEXT H
990 FOR H=12 TO 16 STEP 4
1000 CALL HCHAR(H,12,104)
1010 M(H-4,8)=104

1020 CALL HCHAR(H,20,104)
1030 M(H-4,16)=104
1040 NEXT H
1050 REM $$ INIZIALIZZO A ed B
$$
1060 A=14
1070 B=16
1080 AV=14
1090 BV=16
1100 CALL CHAR(127,"3C7EFFFFFFF
7E3C")
1110 CALL HCHAR(A,B,127)
1120 M(A-4,B-4)=32
1130 FOR X=1 TO NFAN
1140 AP=INT(24*RND)+1
1150 BP=INT(32*RND)+1
1160 CALL GCHAR(AP,BP,CH)
1170 IF CH<>104 THEN 1140
1180 CA(X)=AP
1190 CB(X)=BP
1200 CAV(X)=5
1210 CBV(X)=5
1220 NEXT X
1230 M(1,1)=42
1240 CALL COLOR(13,16,1)
1250 CALL CHAR(121,"3C7E7FE7E7C3
42")
1260 CALL CHAR(126,"0042C3E7E7FF
7E3C")
1270 CALL CHAR(123,"3C7E1F07071F
7E3C")
1280 REM ** SOTTO PROGRAMMA CA
LL KEY **
1290 CALL KEY(1,K,S)
1300 CALL COLOR(9,1,1)
1310 IF FLAG THEN 2720
1320 IF K>5 THEN 1290
1330 K=K+2
1340 ON K GOTO 1350,1370,1290,15
20,1420,1290,1470
1350 CALL HCHAR(A,B,127)
1360 GOTO 1800
1370 A=A-2*(M(A-2,B-4)>0)
1380 GOSUB 1570
1390 CALL HCHAR(AV,BV,43)
1400 CALL HCHAR(A,B,119+K)

1410 GOTO 1800
1420 B=B-2*(M(A-4,B-2)>0)
1430 GOSUB 1570
1440 CALL HCHAR(AV,BV,43)
1450 CALL HCHAR(A,B,119+K)
1460 GOTO 1800
1470 A=A+2*(M(A-6,B-4)>0)
1480 GOSUB 1570
1490 CALL HCHAR(AV,BV,43)
1500 CALL HCHAR(A,B,119+K)
1510 GOTO 1800
1520 B=B+2*(M(A-4,B-6)>0)
1530 GOSUB 1570
1540 CALL HCHAR(AV,BV,43)
1550 CALL HCHAR(A,B,119+K)
1560 GOTO 1800
1570 REM CONTROLLI
1580 IF M(A-4,B-4)<44 THEN 1630

1590 CP=CP+1
1600 IF CP=80 THEN 1720
1610 IF M(A-4,B-4)=96 THEN 1650

1620 CALL SOUND(-90,400,2)
1630 M(A-4,B-4)=43
1640 RETURN
1650 CALL SOUND(-1000,880,3)
1660 M(A-4,B-4)=32
1670 FLAG=1000
1680 VOLTE=9
1690 CASO=1
1700 CALL COLOR(13,14,1)
1710 RETURN
1720 REM FINE QUADRO
1730 CALL HCHAR(AV,BV,32)
1740 CALL HCHAR(A,B,127)
1750 NFAN=NFAN-1*(NFAN<4)
1760 CP=0
1770 CASO=1
1780 GOTO 670
1790 STOP
1800 AV=A
1810 BV=B
1820 GOSUB 2780
1830 CASO=CASO+1
1840 IF CASO<19 THEN 2040
1850 FOR X=1 TO NFAN
1860 LOOP=0
1870 IF LDIFF THEN 2190
1880 CA(X)=CA(X)+2*(CA(X)<A)*(M(
CA(X)-2,CB(X)-4)>42)
1890 CB(X)=CB(X)+2*(CB(X)<B)*(M(
CA(X)-4,CB(X)-2)>42)
1900 CA(X)=CA(X)-2*(CA(X)>A)*(M(
CA(X)-6,CB(X)-4)>42)
1910 CB(X)=CB(X)-2*(CB(X)>B)*(M(
CA(X)-4,CB(X)-6)>42)
1920 GOSUB 1940
1930 GOTO 2010
1940 CALL HCHAR(CAV(X),CBV(X),M(

```

```

CAV(X)-4,CBV(X)-4)
1950 CALL HCHAR(CA(X),CB(X),128)

1960 IF (CB(X)=B)*(CA(X)=A) THEN
2330
1970 CAV(X)=CA(X)
1980 CBV(X)=CB(X)
1990 CALL COLOR(9,2,1)
2000 RETURN
2010 NEXT X
2020 GOTO 1290
2030 REM FANTASMI CASUALI
2040 FOR X=1 TO NFAN
2050 PCP=PCP+1
2060 IF PCP<98 THEN 2080
2070 PCP=1

2080 ON PC(PCP)GOTO 2090,2110,21
30,2150
2090 CA(X)=CA(X)+2*(M(CA(X)-6,CB
(X)-4)>42)
2100 GOTO 2160
2110 CB(X)=CB(X)+2*(M(CA(X)-4,CB
(X)-6)>42)
2120 GOTO 2160
2130 CA(X)=CA(X)-2*(M(CA(X)-2,CB
(X)-4)>42)
2140 GOTO 2160
2150 CB(X)=CB(X)-2*(M(CA(X)-4,CB
(X)-2)>42)
2160 GOSUB 1940
2170 NEXT X
2180 GOTO 1290
2190 REM GESTIONE PER L'INSEG
UIMENTO
2200 BB=BB-1
2210 IF BB THEN 2230
2220 BB=99
2230 ON BIN(BB)GOTO 2240,2280
2240 CA(X)=CA(X)+2*(M(CA(X)-2,CB
(X)-4)>42)*(CA(X)<A)
2250 LOOP=LOOP+1
2260 CA(X)=CA(X)-2*(M(CA(X)-6,CB
(X
-4)>42)*(CA(X)>A)
2270 IF (CA(X)<>CAV(X))+(LOOP>2)
THEN 2320 ELSE 2280
2280 CB(X)=CB(X)+2*(M(CA(X)-4,CB
(X)-2)>42)*(CB(X)<B)
2290 LOOP=LOOP+1
2300 CB(X)=CB(X)-2*(M(CA(X)-4,CB
(X)-6)>42)*(CB(X)>B)
2310 IF (CB(X)<>CBV(X))+(LOOP>2)
THEN 2320 ELSE 2240
2320 GOTO 1920
2330 IF FLAG=0 THEN 2540
2340 IF X=XV THEN 2380
2350 FF=56
2360 XV=X
2370 GOTO 2400

```

```

2380 FF=50
2390 GOTO 2400
2400 FOR LIM=B-1 TO B+1
2410 CALL GCHAR(A,LIM,PIC(LIM))

2420 NEXT LIM
2430 CALL HCHAR(A,B-1,FF)
2440 CALL HCHAR(A,B,48,2)
2450 PUN=PUN+(FF-48)*100
2460 FOR L=1 TO LEN(STR$(PUN))
2470 CALL SOUND(-300,262,2,330,2
,392,2)
2480 CALL HCHAR(3,21+L,ASC(SEG$(
STR$(PUN),L,1)))
2490 NEXT L
2500 FOR LIM=B-1 TO B+1
2510 CALL HCHAR(A,LIM,PIC(LIM))

2520 NEXT LIM

2530 GOTO 1970
2540 FOR H=200 TO 110 STEP -30
2550 CALL SOUND(1000,H,5)
2560 NEXT H
2570 CALL HCHAR(A,B,32)
2580 A=INT(RND*24)+1
2590 B=INT(RND*32)+1
2600 CALL GCHAR(A,B,CHAR)
2610 IF (CHAR<>104)*(CHAR<>43) TH
EN 2580

2620 CASO=4
2630 N_PIC=N_PIC-1
2640 IF N_PIC THEN 1970
2650 CALL CLEAR
2660 PRINT " FINE GIOCO !!!":::
:" PUNTEGGIO :";PUN:::
2670 INPUT " UN' ALTRA PARTITA?
(S/N):
..... ":RISP$
2680 IF (RISP$="S")+(RISP$="s") T
HEN 170
2690 IF (RISP$<>"N")*(RISP$<>"n"
) THEN 2660
2700 CALL CLEAR
2710 STOP

2720 VOLTE=VOLTE-1
2730 IF VOLTE THEN 1320
2740 XV=10000
2750 FLAG=0
2760 CALL COLOR(13,16,1)
2770 GOTO 1320
2780 FOR X=1 TO NFAN
2790 IF (A=CA(X))*(B=CB(X)) THEN
2330
2800 NEXT X
2810 RETURN

```

# Battaglia Navale

Il gioco è simile in tutto e per tutto alla tradizionale battaglia navale che tutti conosciamo. Scopo del gioco è di distruggere la flotta dello Spectrum che consiste, come la vostra, di 1 nave da 5 quadratini, 1 da 4, 3 da 3, 3 da 2, 4 da 1.

Il campo di gioco è costituito da due reticoli. Quello di destra è a vostra disposizione per il posizionamento delle navi. L'operazione si effettua immettendo prima 'o' oppure 'v' a seconda che vogliate la vostra nave in orizzontale o in verticale e successivamente le coordinate del quadratino di partenza. La nave verrà disegnata dal punto di partenza da sinistra verso destra per l'orizzontale oppure dall'alto al basso per il verticale.

```

10 CLS
20 PRINT INK 2;AT 7,12;"© 198
  4";AT 9,14;"by";AT 11,11;"P
  IER CESARE";AT 13,12;"PASQU
  ALI"
30 PRINT FLASH 1;AT 20,4;" FE
  RMA IL REGISTRATORE "
40 PAUSE 300: CLS
50 PRINT INVERSE 1;AT 1,0;" N
  AVI A DISPOSIZIONE "
80 PRINT AT 3,0;"-1x";AT 3
  ,3; INVERSE 1;" 5";AT 5,
  0; INVERSE 0;"-1x";AT 5
  ,3; INVERSE 1;" 4";AT 7,0
  ; INVERSE 0;"-3x";AT 7,
  3; INVERSE 1;" 3";AT 9,0;
  INVERSE 0;"-3x";AT 9,3
  ; INVERSE 1;" 2";AT 11,0;
  INVERSE 0;"-4x";AT 11,
  3; INVERSE 1;"1"
90 PRINT AT 14,0;"Premi qualsi
  asi tasto per          iniziar
  e"
95 PRINT AT 17,0;"Nel corso de
  l programma usa:";AT 19,1;
  INVERSE 1;"C";AT 19,2;
  INVERSE 0;" per colp
  ito";AT 21,1; INVERSE 1;"M"
  ;AT 21,2; INVERSE 0;" p
  er mancato"
100 PAUSE 0: CLS : GO SUB 6000

```

Il reticolo di sinistra nasconde invece le navi dello Spectrum. La mossa d'inizio è assegnata casualmente. Ai colpi del computer si risponde premendo 'm' per mancato, 'c' per colpito. Le coordinate di sparo vanno sempre intese (o digitate, se è il vostro turno) in questo modo: "linea", "colonna".

Il programma prevede che lo Spectrum non ripeta mai i colpi già effettuati e che una volta individuata la nave sparando casualmente, proceda poi sistematicamente alla sua distruzione.

Tutte le istruzioni INPUT sono controllate. In caso di INPUT errato il programma ripete l'INPUT stesso.

```

110 PLOT 8,160: DRAW 104,0:
      DRAW 0,-104: DRAW -104,0:
      DRAW 0,104
112 FOR x=144 TO 232 STEP 8
114 PLOT x,56: DRAW 0,104
115 NEXT x
116 FOR x=64 TO 152 STEP 8
118 PLOT 136,x: DRAW 104,0
119 NEXT x
120 PLOT 136,160: DRAW 104,0:
      DRAW 0,-104: DRAW -104,0:
      DRAW 0,104
122 FOR x=16 TO 104 STEP 8
124 PLOT x,56: DRAW 0,104
125 NEXT x
126 FOR x=64 TO 152 STEP 8
128 PLOT 8,x: DRAW 104,0
129 NEXT x
130 FOR m=97 TO 109
140 PRINT AT 0,m-96;CHR$ m
150 PRINT AT 0,m-80;CHR$ m
160 PRINT AT m-95,15;CHR$ m
170 NEXT m
180 PRINT AT 18,0;"Posiziona le
      tue navi"
185 RESTORE 5500
190 FOR y=1 TO 12
195 READ e: PRINT AT 20,2;"Lung
      hezza nave=";e: IF e=1
      THEN LET a$="v": GO TO 21
      0

```

```

200 INPUT "orizz. o vert. ? (o/
v) ";a$: IF a$<>"v" AND a
$<>"o" THEN GO TO 200
210 INPUT "Posizione di partenz
a ";c$;" ";d$
220 IF CODE c$>109 OR CODE c$<9
7 THEN GO TO 210
230 IF CODE d$>109 OR CODE d$<9
7 THEN GO TO 210
260 IF a$="o" THEN GO TO 275
270 IF a$="v" THEN GO TO 315
275 IF (CODE d$)-81+e>29 THEN
GO TO 200
280 FOR h=0 TO e-1
290 PRINT AT (CODE c$)-95,(
CODE d$)-80+h;"{SG8}";
300 NEXT h
310 GO TO 350
315 IF CODE c$-96+e>14 THEN
GO TO 200
320 FOR h=0 TO e-1
330 PRINT AT (CODE c$)-95+h,(
CODE d$)-80;"{SG8}";
340 NEXT h
350 NEXT y
355 PRINT AT 18,0,,,,,,
360 PRINT AT 18,0;"Attendi un a
ttimo mentre posi- zione l
e mie navi": DIM y(12,11):
DIM x(12)
370 FOR y=1 TO 12: RANDOMIZE 1+
INT (RND*65535)
380 LET x=INT (RND+.5)
390 IF x=0 THEN GO TO 410
400 IF x=1 THEN GO TO 600
410 LET b=1+INT (RND*12+.5):
LET a=2+INT (RND*12+.5)
420 LET x(y)=1+INT (RND*11+.5)
430 FOR h=1 TO y-1
440 IF x(h)=x(y) THEN GO TO 42
0
450 NEXT h
460 RESTORE 2999+x(y)
470 READ a$: LET c=LEN a$
480 IF b+c-1>13 THEN GO TO 410

485 RESTORE 5000
490 FOR v=0 TO c-1
495 READ d

500 LET y(y,d)=a: LET y(y,d+1)=
b+v
510 NEXT v
520 LET y(y,11)=c
530 GO SUB 4000
590 GO TO 800
600 RANDOMIZE 1+INT (RND*65535)
: LET a=2+INT (RND*12+.5):
LET b=1+INT (RND*12+.5)
610 LET x(y)=1+INT (RND*11+.5)

```

```

620 FOR h=1 TO y-1
630 IF x(h)=x(y) THEN GO TO 61
0
640 NEXT h
650 RESTORE 2999+x(y)
660 READ a$: LET c=LEN a$
670 IF a+c-1>14 THEN GO TO 600

680 REM : REM LET y(y,1)=x:
LET y(y,2)=a: LET y(y,3)=b
: LET y(y,4)=c
685 RESTORE 5000
690 FOR v=0 TO c-1
695 READ d
700 LET y(y,d)=a+v: LET y(y,d+1
)=b
710 NEXT v
720 LET y(y,11)=c
730 GO SUB 4000
800 NEXT y
805 DIM z(2000,2): LET p=1:
DIM a(2): LET j=1000:
LET f=0: LET g=0
810 LET x=INT (RND+.5)
820 RESTORE 3020+x
830 READ a$
835 PRINT AT 18,0,,,,,,
840 PRINT AT 18,0;"Sono pronto.
";a$: PAUSE 100
850 IF x=0 THEN GO TO 860
855 IF x=1 THEN GO TO 1000
860 IF g=28 THEN GO TO 3700
870 PRINT AT 18,0,,,,,:
INPUT "Coordinate di sparo
";a$;" ";b$: LET a=CODE a$
-95: LET b=CODE b$-96
872 IF a<2 OR a>14 THEN
GO TO 870
874 IF b<1 OR b>13 THEN
GO TO 870
880 FOR t=1 TO 12
890 FOR u=1 TO 2*y(t,11)-1
STEP 2
900 IF a=y(t,u) AND b=y(t,u+1)
THEN LET f=f+1: PRINT
AT 18,0;"Colpito";AT a,b;"
{SG8}": GO SUB 9000:
PAUSE 25: GO TO 970

910 NEXT u
920 NEXT t
930 PRINT AT 18,0,,,,,;AT 18,0
;"Mancato";AT a,b;"B":
BEEP .1,-10: REM B in modo
G
970 IF f=28 THEN GO TO 3500
980 PAUSE 100: PRINT AT 18,0,,,
,: GO TO j
1000 LET a=97+INT (RND*12+.5):
RANDOMIZE INT (RND*65532)+

```



```

1: LET b=97+INT (RND*12+.5)
: LET aa=a: LET bb=b: LET c
c=1000: GO SUB 8000
1010 PRINT AT 18,0;"Queste sono
le mie coordinate di spar
o: ";CHR$ a;",";CHR$ b
1030 IF INKEY$="m" THEN LET j=1
000: GO TO 860
1040 IF INKEY$="c" THEN LET j=1
060: LET a(1)=a: LET a(2)=b
: LET g=g+1: PRINT OVER 1;
AT a-95,b-80;"A": GO TO 860
: REM A in modo G
1050 GO TO 1030
1060 LET k=a(1): LET m=a(2)
1070 IF k=97 THEN GO TO 1120
1075 LET aa=k-1: LET bb=m: LET c
c=1120: GO SUB 8000
1080 PRINT AT 18,0;"Queste sono
le mie coordinate di spar
o: ";CHR$ (k-1);",";CHR$ m
1090 IF INKEY$="m" THEN LET j=1
120: GO TO 860
1100 IF INKEY$="c" THEN LET j=1
070: LET k=k-1: LET g=g+1:
PRINT OVER 1;AT k-95,m-80
;"A": GO TO 860: REM A in m
odo G
1110 GO TO 1090
1120 LET k=a(1): LET m=a(2)
1130 IF m=109 THEN GO TO 1180
1135 LET aa=k: LET bb=m+1: LET c
c=1180: GO SUB 8000
1140 PRINT AT 18,0;"Queste sono
le mie coordinate di spar
o: ";CHR$ k;",";CHR$ (m+1)
1150 IF INKEY$="c" THEN LET j=1
130: LET m=m+1: LET g=g+1:
PRINT OVER 1;AT k-95,m-80
;"A": GO TO 860: REM A in m
odo G
1160 IF INKEY$="m" THEN LET j=1
180: GO TO 860
1170 GO TO 1150

1180 LET k=a(1): LET m=a(2)
1190 IF k=109 THEN GO TO 1240
1195 LET aa=k+1: LET bb=m: LET c
c=1240: GO SUB 8000
1200 PRINT AT 18,0;"Queste sono
le mie coordinate di spar
o: ";CHR$ (k+1);",";CHR$ m
1210 IF INKEY$="c" THEN LET j=1
190: LET k=k+1: LET g=g+1:
PRINT OVER 1;AT k-95,m-80
;"A": GO TO 860: REM A in m
odo G
1220 IF INKEY$="m" THEN LET j=1
240: GO TO 860
1230 GO TO 1210

```

```

1240 LET k=a(1): LET m=a(2)
1250 IF m=97 THEN GO TO 1000
1255 LET aa=k: LET bb=m-1: LET c
c=1000: GO SUB 8000
1260 PRINT AT 18,0;"Queste sono
le mie coordinate di spar
o: ";CHR$ k;",";CHR$ (m-1)
1270 IF INKEY$="m" THEN LET j=1
000: GO TO 860
1280 IF INKEY$="c" THEN LET j=1
250: LET g=g+1: LET m=m-1:
PRINT OVER 1;AT k-95,m-80
;"A": GO TO 860: REM A in m
odo G
1290 GO TO 1270
3000 DATA " "
3001 DATA " "
3002 DATA " "
3003 DATA " "
3004 DATA " "
3005 DATA " "
3006 DATA " "
3007 DATA " "
3008 DATA " "
3009 DATA " "
3010 DATA " "
3011 DATA " "
3020 DATA "Incomincia pure."
3021 DATA "Incomincio io."
3500 PRINT AT 18,0,,,,,,
3520 PRINT AT 18,0;"Complimenti,
hai vinto!";AT 20,0;"Vuoi g
iocare ancora ? (s/n)"
3530 IF INKEY$="s" THEN CLS :
RUN 70
3540 IF INKEY$="n" THEN STOP
3550 GO TO 3530
3700 PRINT AT 18,0,,,,,,
3710 PRINT AT 18,0;"Ho vinto io.
";AT 20,0;"Vuoi sfidarmi di
nuovo ? (s/n)"
3720 IF INKEY$="n" THEN LET z=0
: GO TO 3750
3730 IF INKEY$="s" THEN LET z=1
: GO TO 3750
3740 GO TO 3720
3750 PRINT AT 18,0;"Questo era i
l mio schema",,,,,,
3760 FOR t=1 TO 12: RESTORE 5000
3770 FOR u=1 TO 2*y(t,11)-1
STEP 2
3780 READ d: PRINT AT y(t,d),y(t
,d+1);"{SG8}"
3790 NEXT u
3800 NEXT t
3810 PAUSE 300
3820 IF z=1 THEN CLS : RUN 70
3830 IF z=0 THEN STOP

```

```

4000 FOR t=1 TO y-1
4010 FOR u=1 TO 2*y(t,1)-1
      STEP 2
4020 FOR s=1 TO 2*y(y,1)-1
      STEP 2
4030 IF y(t,u)=y(y,s) AND y(t,u+
      1)=y(y,s+1) THEN GO TO 380

4040 NEXT s
4050 NEXT u
4060 NEXT t
4070 RETURN
5000 DATA 1,3,5,7,9
5500 DATA 5,4,3,3,3,2,2,2,1,1,1,
      1
6000 FOR i=0 TO 6
6010 POKE USR "a"+i,127
6020 NEXT i
6030 POKE USR "a"+7,0

```

```

6035 RESTORE 7000
6040 FOR i=0 TO 7
6050 READ d: POKE USR "b"+i,d
6060 NEXT i
6070 RETURN
7000 DATA 128,162,148,136,148,16
      2,128,255
8000 FOR q=1 TO p
8010 IF aa=z(q,1) AND bb=z(q,2)
      THEN GO TO cc
8020 NEXT q
8030 LET z(p,1)=aa: LET z(p,2)=b
      b: LET p=p+1
8040 RETURN
9000 RESTORE 9020
9010 FOR r=1 TO 5: READ r1:
      BEEP .1,-r1: NEXT r:
      RETURN
9020 DATA 20,20,40,50,55

```



16K/48K

## Ritorno alla base

Lo scopo di questo gioco è quello di condurre la vostra astronave dall'angolo in basso a sinistra dello schermo sino alla vostra base, posta al limite superiore del video stesso. In questo tentativo incontrerete però due problemi: innanzitutto la porta di accesso alla base si riduce mano a mano che la astronave realizza punteggio avanzando sullo schermo; in secondo luogo incontrerete numerosi ostacoli disseminati lungo il percorso che vi sottrarranno energia fino a farvi esplodere. Il veicolo può spostarsi a sinistra (tasto g), a destra (tasto 0) e in basso (tasto I); una volta avviato il gioco, il movimento in avanti è invece del tutto automatico.

Il punteggio è funzione del movimento

```

1 REM *****
2 REM * RITORNO ALLA BASE *
3 REM *****
5 GO SUB 5050: GO TO 8
8 PAPER 7: CLS
10 LET HI=0
15 LET SCS=0
45 REM istruzioni

```

dell'astronave avanti, a destra e a sinistra (non in basso) e della raccolta di alcuni 'premi' (unità di energia ecc.), che possono farvi guadagnare altre vite (al massimo 9). Di queste ultime avrete particolare bisogno al crescere del livello di difficoltà del gioco: sono infatti predisposti 7 livelli, caratterizzati via via da un numero crescente di ostacoli e da un minor tempo di apertura della porta di accesso alla base. Il programma vi consentirà di provare i differenti livelli prima della partita vera e propria (rispondete con un tasto diverso da 'a' a tale richiesta): in questa fase non verrà però avviata la procedura di verifica del miglior punteggio.

```

50 PRINT AT 9,0; INK 1;
      BRIGHT 1; "{32SG8}R I T O R
      N O A L L A B A S E
      {32SG8}"
52 FOR R=1 TO 20: BEEP .05,R:
      BEEP .01,R+10: BEEP .001,R
      -20: NEXT R

```

```

65 PAUSE 50: CLS : PRINT AT 0,
  0;"PUNTEGGI :
      2,4,6,8,10,12,14 pu
nti a secondadel livello di
difficolta' e per ogni
tuo movimento:avanti
      sinistra
      des
tra Inoltre:"" 50 punti-
per ogni stella(.) e per
ogni stella esplosiva(*)
100 punti- unita' di energ
ia(#)150 punti- per ogni m
issile(↑) 200 punti- rient
ro alla base"
67 PRINT "LIVELLI DI GIOCO:"
"1.....350 secondi
      2.....175 secondi
      3.....113 seco
ndi      4.....85
secondi      5.....
...70 secondi      6
.....58 secondi
      7.....50 secondi"
72 PRINT AT 21,9;"PREMI UN TAS
TO": PAUSE 0: CLS : PRINT
AT 5,0;"COMANDI:
      '0' .....DESTR
A      '9' ...SI
NISTRA      '1'
.....BASSO
MOVIMENTO AVANTI E' AUTOMA
TICO"
75 PRINT AT 21,2;"PREMI UN TAS
TO PER INIZIARE": PAUSE 0:
CLS
77 INPUT "SE DESIDERI LIVELLI
DI GIOCO DETERMINATI AUT
OMATICAMENTE(1/7)PREMI 'a'.
";1$

78 IF 1$="a" THEN LET DI=1:
GO TO 88
80 INPUT "LIVELLO DI DIFFICOLT
A'(1-7)";DI
85 IF DI<1 OR DI>7 THEN
GO TO 80
88 LET SC=0
90 LET L=DI*2
95 IF L>9 THEN LET L=9
102 BORDER 1: PAPER 4: CLS
115 REM allestimento
120 PRINT AT 0,9;"UN ATTIMO....
"
140 FOR Q=0 TO 50*DI
145 LET S=INT (RND*31): LET K=
INT (RND*21+3)
147 IF K>21 THEN LET K=21
150 PRINT AT K,S;"X"
152 BEEP .05,RND*50

```

```

153 IF Q=50*DI THEN GO TO 160
155 PRINT AT 21,0;" "
158 NEXT Q
160 LET R=21: LET C=0
170 PRINT AT RND*17+3,RND*31;"#
"
175 PRINT AT RND*17+3,RND*31;"
"
180 PRINT AT RND*17+3,RND*31;"↑
"
190 PRINT AT RND*17+3,RND*31;"*
"
205 PRINT AT 0,2;"PREMI UN TAST
O PER INIZIARE"
210 PRINT AT 2,0;" I N G R E
S S O B A S E "
235 PRINT AT R,C;"N"
240 PAUSE 0
250 PRINT AT 0,0; INVERSE 1;"
"
290 PRINT AT R,C;" "
295 REM loop di gioco
300 FOR W=1 TO INT 350/DI
302 IF W=INT 350/DI THEN
GO TO 1100
303 PRINT AT R,C;"N"
304 IF L>9 THEN LET L=9
306 REM punteggio/tempo/vite
307 PRINT AT 0,25; INVERSE 1;"V
ITE :";L
309 PRINT AT 0,12;"LIVELLO:";DI
312 PRINT AT 0,0; INVERSE 1;"PU
NTI:";SC

```



```

313 IF HI<SC THEN LET HI=SC
316 PRINT #2; INVERSE 1;"MAX PU
    NT.:";HI
320 REM chiusura casuale porta
323 LET X=INT (RND*10)
325 IF X<2 THEN GO TO 323
330 IF SC>20*X THEN PRINT AT 2
    ,0; INK 2; INVERSE 1;"<<<<"
    ;AT 2,28; INK 2;">>>>"
335 IF SC>30*X THEN PRINT AT 2
    ,4; INK 2; INVERSE 1;"<<<";
    AT 2,25; INK 2;">>>"
340 IF SC>90*X THEN PRINT AT 2
    ,7; INK 2; INVERSE 1;"<<";
    AT 2,23; INK 2;">>"
342 IF SC>120*X THEN PRINT
    AT 2,9; INK 2; INVERSE 1;"<
    ";AT 2,22; INK 2;">"
343 IF SC>180*X THEN PRINT
    AT 2,10; INK 2; INVERSE 1;"
    <<<";AT 2,19; INK 2;">>>"
344 IF SC>220*X THEN PRINT
    AT 2,13; INK 2; INVERSE 1;"
    <<";AT 2,16; INK 2;">>>"
345 IF SC>280*X THEN PRINT
    AT 2,0; INVERSE 1;"          T
    ROPPO TARDI          ":
    PAUSE 200: GO TO 3000
348 PRINT AT R,C;"_"
349 REM movimento
350 IF INKEY$="9" THEN LET C=C
    -1: LET SC=SC+(DI*2)
355 IF INKEY$="1" THEN LET R=R
    +1
360 IF INKEY$="0" THEN LET C=C
    +1: LET SC=SC+(DI*2)
362 IF INKEY$="" THEN LET R=R-
    1: LET SC=SC+(DI*2)
365 IF R>20 THEN LET R=21
366 IF C<0 THEN LET C=0
367 IF C>30 THEN LET C=31
369 REM controllo posizione
370 IF SCREEN$ (R,C)="#" THEN
    GO TO 7000
371 IF SCREEN$ (R,C)="." THEN
    GO TO 7100
372 IF SCREEN$ (R,C)="↑" THEN
    GO TO 7200

```

```

373 IF SCREEN$ (R,C)="*" THEN
    GO TO 7300
374 IF SCREEN$ (R,C)="X" THEN
    GO TO 7400
375 IF SCREEN$ (R,C)("<" THEN
    GO TO 1000
376 IF SCREEN$ (R,C)(">" THEN
    GO TO 1000
377 PRINT AT 1,18; INVERSE 1;"T
    EMPO : ";W
379 IF R=2 THEN GO TO 2000
380 REM disegno casuale durante
    il gioco
381 LET X=INT (RND*31)
385 LET Z=INT (RND*21)
390 IF Z<3 THEN GO TO 380
392 PRINT AT Z,X;"X"
395 IF R<2 THEN LET R=2
500 NEXT W
1000 REM vite perse/ urti/tempo
    trascorso
1001 PRINT AT 10,9; FLASH 1;"SEI
    MORTO !!!"
1005 FOR A=0 TO -20 STEP -1:
    BEEP .05,A: NEXT A
1010 FOR f=0 TO 200: NEXT f:
    CLS : PRINT AT 10,9;"UN'AL
    TRA ?(S/N)"
1015 INPUT A$
1018 IF HI<SCS THEN LET HI=SCS
1020 IF A$="S" OR A$="s" THEN
    LET SCS=0: CLS : GO TO 77
1030 STOP
1100 PRINT AT 10,6; FLASH 1;"IL
    TEMPO E' SCADUTO !!"
1110 FOR A=0 TO 25: BEEP .01,A-1
    0: BEEP .01,A+10: NEXT A
1115 GO TO 1010
2000 REM ritorno alla base
2001 LET SC=SC+200: PRINT AT 10,
    8; BRIGHT 1;"CE L'HAI FATTA
    !!!"
2002 FOR A=1 TO 3: FOR S=1 TO 20
2005 BORDER RND*6: PAPER RND*6
2010 BEEP .05,S: NEXT S: NEXT A
2015 CLS : LET SCS=SCS+SC
2018 IF HI<SCS THEN LET HI=SCS
2020 IF I$("<>"a" THEN GO TO 20
    35
2025 LET DI=DI+1: IF DI<=7
    THEN GO TO 2035
2030 PRINT AT 10,7; PAPER 7;
    INK 4; FLASH 1;" SEI IL
    MIGLIORE !! ": PAUSE 50:
    PRINT AT 15,6;"Punteggio f
    inale : ";SCS: LET SCS=0:
    PAUSE 200: GO TO 1010
2035 PRINT AT 10,8; PAPER 7;
    FLASH 1;"Punti : ";SCS:
    PAUSE 200

```

**PAPER**  
soft

ritaglia  
e  
conserva  
ti  
attendono  
grosse  
sorprese!

```

2045 LET SC=0
2050 PAPER 7: CLS : IF 1$="a"
      THEN GO TO 90
2060 GO TO 80
3000 REM sparizione della base
3001 PRINT AT 10,0; FLASH 1;"TRO
      PPO PIANO, LA BASE E' SVANI
      TA"
3010 FOR A=1 TO 30: BEEP .01,A+2
      : BEEP .01,A-30: NEXT A
3020 GO TO 1010
5040 REM UDG-astronave (N)
5050 FOR L=0 TO 7
5053 DATA 24,60,126,255,126,60,2
      4,0

```

```

5055 READ N: POKE USR "N"+L,N
5060 NEXT L
6000 RETURN
7000 LET SC=SC+100: LET L=L+1:
      BEEP .01,30: GO TO 371
7100 LET SC=SC+50: LET L=L+1:
      BEEP .01,20: GO TO 372
7200 LET SC=SC+150: LET L=L+1:
      BEEP .01,10: GO TO 373
7300 LET SC=SC+50: LET L=L+1:
      BEEP .01,40: GO TO 374
7400 LET L=L-1: BEEP .01,50:
      IF L=0 THEN GO TO 1000
7401 GO TO 375

```



16K/48K

## Centipede

*In questo gioco, che metterà a dura prova la saldezza dei vostri nervi, l'obiettivo è quello di far avanzare il più possibile il centipede nel*

*prato senza che esso tocchi i fiori velenosi in esso presenti. Non è così facile come sembra; provatelo e... buon viaggio!*

```

10 FOR i=0 TO 7: READ a:
      POKE USR "a"+i,a: NEXT i:
      DATA 60,126,255,153,255,12
      6,60,66
90 LET h=0: LET a=16: LET s=0
100 BORDER 1: PAPER 1: INK 6
130 CLS : PRINT AT 0,11;
      FLASH 1;"CENTIPEDE";AT 4,0
      ; FLASH 0;"Il tuo centipede
      deve evitare tutti i fio
      ri velenosi, o ";AT 6,11;
      FLASH 1;"MORIRA'!"
140 PRINT AT 13,0; INK 4;"CONTR
      OLLI:"'"5= SINISTRA"'8=DE
      STRA": PRINT #0; FLASH 1;"
      SE SEI PRONTO, PREMI UN TAS
      TO ": PAUSE 0
150 FOR f=0 TO 48 STEP .3:
      BEEP .01,f: NEXT f: CLS
160 PRINT AT 7,a; INK 4;
      BRIGHT 1;"A"
172 FOR i=1 TO 5: PRINT AT 21,
      RND*31;"*": NEXT i
180 LET a=a+(INKEY$="8" AND a<3
      1)-(INKEY$="5" AND a>0)
190 LET l=USR 3280: IF
      SCREEN$ (7,a)="*" THEN
      GO SUB 500
210 BEEP .001,20: LET s=s+.5:

```

```

      GO TO 160
500 FOR f=48 TO 0 STEP -.3:
      BEEP .01,f: NEXT f: CLS :
      PRINT AT 0,10;"RECORD ";h;
      AT 5,7;"PUNTEGGIO ";s;AT 10
      ,0;" Sei davvero un po' sc
      arso... "
510 IF s>h THEN LET h=s:
      PRINT AT 15,0; INK 2;
      PAPER 6; BRIGHT 1; FLASH 1
      ;"...anche se e' il NUOVO R
      ECORD!"
515 PRINT AT 21,2; INK 4;
      FLASH 1; BRIGHT 1;"Per rif
      arti, premi un tasto":
      LET f=0
516 BEEP .1,20: PRINT AT 7,f;
      INK 4;"A";AT 7,f-5;" ":
      LET f=f+1: IF f>30 THEN
      LET f=0: FOR g=25 TO 30:
      BEEP .01,g: PRINT AT 7,g;"
      ";: NEXT g
518 IF INKEY$="" THEN GO TO 51
      6
520 FOR f=0 TO 21: LET l=USR 32
      80: BEEP .05,f: NEXT f
530 FOR f=0 TO 10: NEXT f:
      BEEP 1,22: LET s=0: LET a=.
      16: RETURN

```

# Mosaic puzzle

C'è una tavoletta quadrata con 15 caselle numerate, anch'esse quadrate, libere di scorrere una accanto all'altra, e uno spazio vuoto al posto della sedicesima: bisogna metterle in ordine nel minor tempo possibile. In pratica è l'antenato del Cubo di Rubik. Risolverate questo rompicapo grazie al vostro C64.

Potete fissare un limite di tempo, se volete; il movimento delle caselle, che sono contras-

```

100 POKE53280,0:POKE53281,0:POKE
    55,176:POKE56,29:CLR:POKE542
    76,8:POKE54283,8
110 POKE54277,0:POKE54278,255:PO
    KE54284,0:POKE54285,255:POKE
    54296,15
120 S=1355:SC=S+54272:DIMAS(16)
130 PRINT"{CLR}":G=1632:X=0:DX=1
    :P=55904:S1=54276:S2=54283:A
    D=1232:R=14
140 PRINT"{CLR}{GIU'}"TAB(11)" M
    OSAIC PUZZLE"TAB(50)"
    [< 17 Y>]{GIU'}"
150 :
160 :
170 REM
180 PRINT"[<7>] VUOI UN LIMITE D
    I TEMPO ? ";:GOSUB270
190 IFIN$<>"S"THEN240
200 H=1:INPUT"{HOME}{ 6 GIU'} OR
    E{ 3 SPAZI}MIN{ 2 SPAZI}SEC
    { 2 SPAZI}(6 NUMERI)";T$:IFL
    EN(T$)<>6THEN200
210 IFLEFT$(T$,2)>"23"ORLEFT$(T$
    ,2)<"0"THEN200
220 IFMID$(T$,3,2)>"59"ORMID$(T$
    ,3,2)<"0"THEN200
230 IFRIGHT$(T$,2)>"59"ORRIGHT$(
    T$,2)<"0"THEN200
240 PRINT"{GIU'} TASTO PER SU: "
    ;:GOSUB270:U$=IN$:PRINT"
    {GIU'} PER GIU : ";:GOSUB270
    :D$=IN$
250 PRINT"{GIU'} PER SIN.: ";:GO
    SUB270:L$=IN$:PRINT"{GIU'} P
    ER DES. : ";:GOSUB270:R$=IN$
260 GOTO310
270 PRINT"[<X>";:WAIT198,255:GE
    TIN$:PRINT"{SIN}";:POKE216,1
    :PRINTIN$:RETURN

```

segnate dalle cifre da 1 a 9 e dalle lettere da a F, si ottiene mediante i tasti E (su), X (giù), F (destra) e A (sinistra). Premete RETURN per sospendere il gioco.

Prima di ogni partita potete scegliere se riportare le caselle nella configurazione iniziale, oppure riprendere dalla posizione in cui erano rimaste dopo l'ultima partita giocata.

```

280 :
290 :
300 REM
310 PRINT"{CLR}"TAB(43)"DEFINISC
    I LA CONFIGURAZIONE"
320 PRINT"{GIU'}{ 3 SPAZI}1 2 3
    4 5 6 7 8 9"SPC(23)"A B C D
    E F {RVS}SPAZIO"
330 PRINTTAB(5)"{GIU'} IN QUALUNQ
    UE ORDINE":PRINTTAB(248)"OBB
    IETTIVO"
340 FORK=0TO3:POKE1592+K,100:POK
    E1792+K,99:POKE55864+K,R:POK
    E56064+K,R
350 POKE1631+K*40,103:POKE1636+K
    *40,101:POKE55903+K*40,R:POK
    E55908+K*40,R:NEXT
360 FORI=1TO16:POKEG+X,63:POKEP+
    X,1
370 WAIT198,255:GETAS(I):FORL=I-
    1TO0STEP-1:IFAS(I)=AS(L)THEN
    370
380 NEXT:IFAS(I)=" "THENFORK=0TO
    4:POKE55471+K,15:NEXT:B2=32:
    GOTO420
390 IF(AS(I)<"1"ORAS(I)>"F")OR(A
    $(I)>"9"ANDAS(I)<"A")THEN370
400 B=VAL(AS(I)):B2=B+48:IFBTHEN
    POKE55477+2*B,15:GOTO420
410 B=ASC(AS(I))-64:B2=B:POKE554
    57+2*B,15
420 POKEG+X,B2:X=X+DX:IFX=4THENG
    =G+40:P=P+40:X=0
430 NEXT
440 :
450 :
460 REM
470 PRINT"{HOME}":FORI=0TO64:PRI
    NT"{ 4 SPAZI}";:NEXT:PRINT"

```

```

{HOME}"TAB(127)"PUZZLE"
480 FORK=0TO3:POKE1192+K,100:POK
E55464+K,R:POKE1392+K,99:POK
E55664+K,R
490 POKE1231+K*40,103:POKE55503+
K*40,R:POKE1236+K*40,101:POK
E55508+K*40,R:NEXT
500 READA,B,C:IFA>=0THENPOKEAD+A
,B:POKE55504+A,C:GOTO500
510 FORI=1TO500:NEXT:POKES1-3,80
:POKES1,33:PRINT"{HOME}"TAB(
28)"{ 10 GIU' }{RED}{WHT}!GO!
[<7>]"
520 FORT=1TO300:NEXT:PRINT"
{HOME}"TAB(28)"{ 10 GIU' }
{ 4 SPAZI}":POKES1,8:TI$="00
0000"
530 PRINT"{HOME}"TAB(25)"LIMITE:
{CYN}";:IFT$=""THENPRINT"NES
SUNO":GOTO580
540 PRINTLEFT$(T$,2)": "MID$(T$,3
,2)": "RIGHT$(T$,2)" [<7>]"
550 :
560 :
570 REM
580 PRINT"{HOME}TEMPO SPESO :
{WHT}"LEFT$(TI$,2)": "MID$(TI
$,3,2)": "RIGHT$(TI$,2)" [<7>]"
"
590 IFH=1ANDT$<=TI$THEN750
600 GETB$:J=31-PEEK(56320)AND31:
IFB$=""ANDJ=0THEN580
610 IFB$=CHR$(13)ORJ=16THENWV=0:
GOTO780
620 IFB$=D$OR(JAND2) THENDR=-40:C
K=100:GOTO660
630 IFB$=L$OR(JAND4) THENDR=1:CK=
101:GOTO660
640 IFB$=R$OR(JAND8) THENDR=-1:CK
=103:GOTO660
650 DR=40:CK=99:IFB$<>U$AND(JAND
1)=0THEN580
660 IFPEEK(S+DR)=CKTHEN580
670 POKES,PEEK(S+DR):POKESC,PEEK
(SC+DR):POKES+DR,32:S=S+DR:S
C=SC+DR
680 FORM=0TO120STEP40:FORN=0TO3:
W=PEEK(AD+M+N)AND127:IFW<>PE
EK(1632+M+N) THEN580

```

```

690 NEXT:NEXT:PRINT"{HOME}"TAB(2
4)"{ 5 GIU' }{CYN}{RVS}HAI VI
NTO! [<7>]":POKES1-3,0:POKES1
,33:WN=1
700 READN1,N2,D:IFN1=-1THENPOKES
1,8:GOTO780
710 POKES1-4,N1:POKES1-3,N2:FORT
=1TOD:NEXT:GOTO700
720 :
730 :
740 REM
750 PRINT"{HOME}"TAB(23)"
{ 5 GIU' }{WHT}{RVS}HAI PERSO
! [<7>]":POKES1-3,10:POKES1,1
7:WN=1
760 POKES2-3,60:POKES2,129:FORT=
1TO300:NEXT:POKES2,8:POKES1,
8
770 :
780 TM$=TI$:PRINT"{HOME}"TAB(21)
"{ 9 GIU' }(1) RIGIOCARE"
790 PRINTTAB(21)"{GIU' }(2) FINIR
E"
800 GETV$:IFV$<"1"ORV$>"2"THEN80
0
810 IFV$="1"THENRUN
820 IFV$="2"THENEND
830 IFWVTHEN800
840 PRINT"{HOME}{ 8 GIU' }":FORI=
1TO6:PRINTTAB(21)"
{ 18 SPAZI}":NEXT
850 TI$=TM$:GOTO580
860 :
870 :
880 REM MUSICA
890 DATA0,49,1,1,178,3,2,51,1,3,
180,3
900 DATA40,53,1,41,182,3,42,55,1
,43
910 DATA184,3,80,57,1,81,129,3,8
2,2,1
920 DATA83,131,3,120,4,1,121,133
,3,122
930 DATA6,1,123,32,3,-1,-1,-1
940 DATA 96,22,150,0,0,50,96,22,
75,0,0,50,96,22,75,49,28,175
,96,22,115,49,28
950 DATA175,135,33,250,0,0,0,-1,
-1,-1

```

## Campo minato

Siete alla guida di un camion con una squadra di artificieri ed il vostro compito è quello

di disinnescare tutte le bombe presenti in una serie di campi minati. All'inizio il vostro



Joystick (opzionale per il C64)

camion appare al centro del campo, dove le mine sono rappresentate con delle "x" e le bombe con dei cerchietti; il movimento del camion è comandato dal joystick. Dopo un po' di tempo le bombe, una per volta, appaiono in negativo: ciò significa che sono innescate. Successivamente iniziano a lampeggiare, il che indica che esploderanno da un momento all'altro. Per neutralizzarle dovete passarci sopra, e in tal modo ottenete 10 punti per le bombe normali, 20 per quelle in negativo e 30 per quelle lampeggianti. Ogni volta che una bomba esplose, distrugge ciò

che si trova nei dintorni (eventualmente anche voi) e fa scoppiare le bombe che le stanno vicino.

Mentre vi spostate nel campo, evitate ovviamente le mine ma anche i crateri lasciati dalle bombe esplose (asterischi). Quando un campo resta privo di bombe, sia perché sono scoppiate sia perché voi le avete neutralizzate, lo schermo è completato e si passa al successivo, con un maggior numero di bombe. Alla fine di ogni schermo sarete penalizzati di 10 punti per ogni bomba esplosa.

```

45 POKE53280,0:POKE53281,0
50 GOSUB1130
60 :
70 DIMBT(37),B3(37),B4(37),BP(37)
   ),BS(37),XM(4),YM(4),BC(25)
80 DEFFNY(X)=INT((X-1024)/40)
90 DEFFNX(X)=(X-40*FNY(X))-1024
100 DEFFNS(X)=1024+PX+40*PY
110 DEFFNP(X)=1307+INT(34*RND(1))
   )+40*INT(15*RND(1))
120 DEFFNN(X)=PEEK(FNS(X))
130 FORJ=1TO4:READXM(J),YM(J):NEXT
140 DATA0,-1,0,1,-1,0,1,0
150 SC=0:BT=1680:NB=4:NW=0:D=54272
160 PRINT"{CLR}";:POKE53272,21
170 PRINT"{RVS}{WHT}CAMPO***_ PU
   NTI: 0"
180 PRINT"{RVS}{WHT}*****_ RE
   CORD: ";HS
190 PRINT"{RVS}{WHT}**MINATO_ SC
   HERMO: 1"
200 PRINT"{RVS}{WHT}{ 8 SPAZI}_
   {DES}";:IFNL<>1THENFORJ=1TON
   L-1:PRINT"Z";:NEXT
210 FORJ=1024TO1183:IFPEEK(J)=32
   THENPOKEJ,160:POKEJ+D,1
220 NEXT
230 XP$="{DES}{RED}U-I{GIU'}
   { 4 SIN}{ 2 U}_-{ 2 I}{GIU'}
   { 6 SIN}{ 3 U}_-{ 3 I}{GIU'}
   { 7 SIN}{ 3 *}_*{ 3 *}{GIU'}
   { 7 SIN}{ 3 J}_-{ 3 K}"
235 XP$=XP$+"{RED}{GIU'}
   { 6 SIN}{ 2 J}_-{ 2 K}{GIU'}
   { 4 SIN}J-K"
240 S$="{HOME}{ 24 GIU' }"

```

```

250 QS="{WHT}{ 40 DES}"
260 XR$="{DES}{WHT}{ 3 SPAZI}
   {GIU'}{ 4 SIN}{ 5 SPAZI}
   {GIU'}{ 6 SIN}{ 7 SPAZI}
   {GIU'}{ 7 SIN}{ 3 SPAZI}*
   { 3 SPAZI}{GIU'}{ 7 SIN}
   { 7 SPAZI}"
265 XR$=XR$+"{GIU'}{ 6 SIN}
   { 5 SPAZI}{GIU'}{ 4 SIN}
   { 3 SPAZI}"
270 :
280 BG=0:NW=NW+1:IFNW>11THEN310
290 NB=NB+1.5:IFNW=1THEN330
300 IFNW<6THENBT=BT-180
310 PRINT"{HOME}{ 2 GIU'}{RVS}";
   TAB(18);NW
320 POKEFNS(1),32:FORJ=1TONB:POK
   EBP(J),32:NEXT
325 FORJ=1TO25:POKEBC(J),32:NEXT
330 BN=INT(NB):FORJ=1TONB:BS(J)=
   1:NEXT
340 FORJ=1TONB
350 BT(J)=(.4+INT(61*RND(1)))/100
   )*BT
360 B3(J)=BT(J)+.5*BT(J):B4(J)=B
   3(J)+BT(J)
370 NEXT
380 PX=19:PY=15:POKEFNS(1),90:PO
   KEFNS(1)+D,1
390 FORJ=1TONB
400 BP(J)=FNP(1):IFPEEK(BP(J))<>
   32THEN400
410 POKEBP(J),87:POKEBP(J)+D,8:N
   EXT:NN=0
415 FORJ=1TO25
416 BC(J)=FNP(1):IFPEEK(BC(J))<>
   32THEN416

```



```

417 IFPEEK(BC(J)+1)=87THEN416
419 POKEBC(J),86:POKEBC(J)+D,5:N
EXT
420 GETR$:IFR$<>" "THEN420
430 DR=0:TX=TI
440 :
450 R=(15-(PEEK(56321)AND15))*2
460 IFR<>0THENDR=LOG(R)/LOG(2)
470 IFR=0THEN490
480 :
490 IFDR=0THEN600
500 POKEFNS(1),32:PX=PX+XM(DR):P
Y=PY+YM(DR)
510 IFPX<0THENPX=39
520 IFPX>39THENPX=0
530 IFPY<4THENPY=24
540 IFPY>24THENPY=4
550 X=FNN(1)
560 IFX=32THENPOKEFNS(1),90:POKE
FNS(1)+D,1:GOTO600
570 IFX=42ORX=86THEN960
580 GOTO890
590 :
600 NN=NN+1:IFNN>INT(NB)THENNN=1

610 IFBS(NN)=0THEN600
620 TG=TI-TX
630 IFTG>B4(NN)THENN1=NN:GOTO720

640 IFBS(NN)>2THEN690
650 IFTG>BT(NN)THENBS(NN)=2
660 IFTG>B3(NN)THENBS(NN)=3
670 IFBS(NN)=1THEN450
680 IFBS(NN)=2THENPOKEBP(NN),215
:POKEBP(NN)+D,1:GOTO450
690 IFBS(NN)=3THENPOKEBP(NN),87:
POKEBP(NN)+D,1:BS(NN)=4:GOTO
450
700 IFBS(NN)=4THENPOKEBP(NN),215
:POKEBP(NN)+D,1:BS(NN)=3:GOT
O450
710 :
720 TQ=TI:PD=0
725 X$="{OFF}" +LEFT$(S$,FNY(BP(N
1)))-2)+LEFT$(Q$,FNN(BP(N1))-
1)
730 BS(N1)=0:N2=0:PRINTX$;XP$;
740 FORJ=1TONB:X=PEEK(BP(J)):IFB
S(J)=0THEN760
750 IFX<>87ANDX<>215ANDX<>218THE
NN2=J
760 NEXT:IFFNN(1)<>90ANDFNN(1)<>
218THENPD=1
770 PRINTX$;XR$;:GR=129:GOSUB200
0
780 FORJ=1TONB:IFPEEK(BP(J))=32A
NDBS(J)<>0THENPOKEBP(J),87-1
28*(BS(J)>1)
790 NEXT:BN=BN-1

```

```

800 IFPD=1THEN960
810 IFBN=0THEN840
820 IFN2=0THENTX=TX+(TI-TQ):GOTO
450
830 N1=N2:GOTO725
840 PRINT"{HOME}{ 2 GIU' }{RVS}";
TAB(20);
850 FORJ=1TO20:PRINT"{RVS}
{ 2 SPAZI}COMPLETATO
{ 12 SIN}";:FORK=1TO100:NEXT
860 PRINT"{RVS}{ 12 SPAZI}
{ 12 SIN}";:FORK=1TO100:NEXT
:NEXT
870 SC=SC-10*(INT(NB)-BG):IFSC<0
THENSC=0
880 PRINT"{ 4 SIN}{ 3 SU}
{ 10 SPAZI}{ 10 SIN}";SC:GOT
O280
885 :
890 BG=BG+1:TQ=TI:POKEFNS(1),218

895 FORJ=1TONB:IFPEEK(BP(J))=218
THENAJ=BS(J):BS(J)=0
900 NEXT
910 IFAJ=4THENAJ=3
920 SC=SC+10*AJ:PRINT"{HOME}
{RVS}";TAB(16);SC
930 GR=33:GOSUB2000:BN=BN-1:IFBN
=0THEN840
940 TX=TX+(TI-TQ):GOTO450
950 :
960 GR=129:GOSUB2000
961 TQ=TI:FORJ=1TO20:POKEFNS(1),
42:FORK=1TO25:NEXT:POKEFNS(1
),170
970 FORK=1TO25:NEXT:NEXT:POKEFNS
(1),32:NL=NL-1
980 POKE1153+NL,160:DR=0:PX=19:P
Y=15
990 IFNL=0THEN1045
1000 IFBN=0THEN840
1010 GETR$:IFR$<>" "THEN1010
1020 FORJ=1TONB:IFPEEK(BP(J))=32
ANDBS(J)<>0THENPOKEBP(J),87
-128*(BS(J)>1)
1030 NEXT
1040 POKEFNS(1),90:TX=TX+(TI-TQ)
:GOTO450
1045 IFSC>HSTHENHS=SC:PRINT"
{HOME}{GIU' }{RVS}";TAB(17);
HS
1050 FORJ=1TO1500:NEXT:PRINT"
{HOME}{WHT}{ 3 GIU' }{RVS}";
TAB(24);"FINE GIOCO{GIU' }
{ 12 SIN} ANCORA?";
1060 PRINT"(S/N) { 5 SIN}";
1080 PRINT"{RVS}S/{OFF}N
{ 3 SIN}";

```

```

1081 FORJ=1TO99:NEXT
1082 PRINT"{OFF}S{RVS}/N
      { 3 SIN}";
1083 FORJ=1TO99:NEXT
1084 GETR$:IFR$="S"THEN1110
1090 IFR$<>"N"THEN1080
1100 PRINT"{CLR}{WHT}{ 2 SPAZI}
      CIAO!{ 2 SPAZI}":END
1110 GOSUB1130:GOTO150
1120 :
1130 PRINT"{CLR}{RVS}{WHT}
      { 6 DES}{ 2 SPAZI}C A M P
      O{ 4 SPAZI}M I N A T O
      { 2 SPAZI}":POKE53272,23
1140 PRINT"{ 2 GIU'}VUOI LE ISTR
      UZIONI? (S/N)"
1150 GETR$:IFR$="N"THEN1410
1160 IFR$<>"S"THEN1150
1180 PRINT"{CLR}LO SCOPO DEL GIO
      CO E' DISINNESARE"
1190 PRINT"PIU' BOMBE POSSIBILI
      PRIMA CHE QUESTE"
1200 PRINT"SCOPPINO. PER FARE CI
      O' DEVI RAGGIUN-"
1210 PRINT"GERLE CON IL TUO CAMI
      ON."
1220 PRINT"LE BOMBE DOPO POCO TE
      MPO APPAIONO IN"
1230 PRINT"NEGATIVO, POI LAMPEGG
      IANO; A QUESTO"
1240 PRINT"PUNTO STANNO PER ESPL
      ODERE: ATTENTO!"
1250 PRINT"SE UNA BOMBA ESPLODER
      A' VICINO A TE,"
1260 PRINT"SALTERAI IN ARIA! UNA
      ESPLOSIONE PUO'"
1270 PRINT"ANCHE PROVOCARE UNA R
      EAZIONE A CATENA."
1280 PRINT"NON FINIRE NEI CRATER
      I DELLE BOMBE (*)"
1290 PRINT"O CONTRO UNA MINA (X
      ), SE NON VUOI"

1300 PRINT"ESSERE DISTRUTTO."
1320 PRINT"I COMANDI SONO:
      { 3 SPAZI}{RVS}1{OFF}
      { 4 SPAZI}SU"
1330 PRINT"{ 18 SPAZI}{RVS}CTRL
      {OFF} SINISTRA"
1340 PRINT"{ 18 SPAZI}{RVS}2
      {OFF}{ 4 SPAZI}DESTRA"
1350 PRINT"{ 18 SPAZI}{RVS}+
      {OFF}{ 4 SPAZI}GIU'"
1355 PRINT"OPPURE IL JOYSTICK (P
      ORTA 1).".
1360 PRINT"UNA VOLTA PARTITO, IL
      TUO CAMION NON"
1370 PRINT"PUO' FERMARSI, MA PUO
      ' ATTRAVERSARE IL"
1380 PRINT"BORDO DELLO SCHERMO E
      RICOMPARIRE SUL"
1385 PRINT"LATO OPPOSTO."
1390 PRINT"{GIU'}PREMI RETURN PE
      R CONTINUARE."
1400 GETR$:IFR$<>CHR$(13)THEN140
      0
1410 PRINT"{CLR}LIVELLO DI DIFFI
      COLTA' ? (0-3)"
1420 GETR$:IFR$<"0"ORR$>"3"THEN1
      420
1430 NL=4-VAL(R$):RETURN
1900 END
2000 :
2010 QW=54272
2020 FORS=QWTOQW+24:POKES,0:NEXT

2025 POKEQW+24,47
2030 POKEQW+5,71:POKEQW+6,240
2050 POKEQW+4,129:POKEQW+1,36:PO
      KEQW,85
2060 FORT=1TO250:NEXT
2070 FORT=15TO0STEP-1:POKEQW+24,
      T:NEXT
2080 RETURN
10000 PRINTPEEK(56321):GOTO10000

```



## Campo minato

exp. 3K byte

Siete alla guida di un camion con una squadra di artificieri ed il vostro compito è quello di disinnescare tutte le bombe presenti in una serie di campi minati. All'inizio il vostro camion appare al centro del campo, dove le mine sono rappresentate con delle "x" e le bombe con dei cerchietti; il movimento del

camion è comandato dal joystick. Dopo un po' di tempo le bombe, una per volta, appaiono in negativo: ciò significa che sono innescate. Successivamente iniziano a lampeggiare, il che indica che esploderanno da un momento all'altro. Per neutralizzarle dovete passarci sopra, e in tal modo ottenete 10

punti per le bombe normali, 20 per quelle in negativo e 30 per quelle lampeggianti. Ogni volta che una bomba esplode, distrugge ciò che si trova nei dintorni (eventualmente anche voi) e fa scoppiare le bombe che le stanno vicino.

Mentre vi spostate nel campo, evitate ovviamente le mine ma anche i crateri lasciati

dalle bombe esplose (asterischi). Quando un campo resta privo di bombe, sia perché sono scoppiate sia perché voi le avete neutralizzate, lo schermo è completato e si passa al successivo, con un maggior numero di bombe. Alla fine di ogni schermo sarete penalizzati di 10 punti per ogni bomba esplosa.

```

45 POKE36879,8
50 GOSUB1410
60 :
70 DIMBT(37),B3(37),B4(37),BP(37),
  BS(37),XM(4),YM(4),BC(25)
80 DEFFNY(X)=INT((X-FNV(X))/22)
85 DEFFNV(X)=4*(PEEK(36866)AND128)+64*(PEEK(36869)AND120)
90 DEFFNX(X)=(X-22*FNY(X))-FNV(X)
95 DEFFNC(X)=37888+4*(PEEK(36866)AND128)
98 DEFFND(X)=FNC(X)-FNV(X)
100 DEFFNS(X)=FNV(X)+PX+22*PY
110 DEFFNP(X)=FNV(X)+178+INT(18*RND(1))+22*INT(11*RND(1))
120 DEFFNN(X)=PEEK(FNS(X))
130 FORJ=1TO4:READXM(J),YM(J):NEXT
140 DATA1,0,-1,0,0,-1,0,1
150 SC=0:BT=(FNV(1)+360):NB=4:NW=0
160 PRINT"{CLR}"CHR$(142);
170 PRINT"{RVS}{WHT}CAMPO***_PUNTI: 0"
180 PRINT"{RVS}{WHT}*****_RECORD:"HS;
190 PRINT"{HOME}{ 2 GIU' }{RVS}**MINATO-SCHERMO:{ 2 SPAZI}1"
200 PRINT"{RVS}{WHT}{ 8 SPAZI}_{DES}";:IFNL<>1THENFORJ=1TONL-1:PRINT"Z";:NEXT
210 FORJ=FNV(1)TOFNV(1)+109:IFPEEK(J)=32THENPOKEJ,160:POKEJ+FND(1),1
220 NEXT
230 XP$="{RED}{ 2 GIU' }{ 8 DES}{ 7 SIN}U-I{GIU' }{ 7 SIN}{ 4 DES}{SIN} ***{GIU' }{ 3 SIN}J-K"
240 S$="{HOME}{ 22 GIU' }"
250 Q$="{WHT}{ 22 DES}"
260 XR$="{GIU' }{ 5 SPAZI }{GIU' }{ 6 SIN }{ 7 SPAZI }{GIU' }{ 7 SIN }{ 3 SPAZI }*{ 2 SPAZI }{GIU' }{ 5 SIN }{ 8 SPAZI }"
270 :

```

```

280 BG=0:NW=NW+1:IFNW>11THEN310
290 NB=NB+1.5:IFNW=1THEN330
300 IFNW<6THENBT=BT-99
310 PRINT"{HOME}{ 2 GIU' }{RVS}"TAB(18)NW
320 POKEFNS(1),32:FORJ=1TONB:POKEBP(J),32:NEXT
325 FORJ=1TO25:POKEBC(J),32:NEXT
330 BN=INT(NB):FORJ=1TONB:BS(J)=1:NEXT
340 FORJ=1TONB
350 BT(J)=((.4+INT(61*RND(1)))/100)/.75)*BT
360 B3(J)=BT(J)+.3*BT(J):B4(J)=B3(J)+.2*BT(J)
370 NEXT
380 PX=11:PY=11:POKEFNS(1),90:POKE(FNS(1)+FND(1)),1
390 FORJ=1TONB
400 BP(J)=FNP(1):IFPEEK(BP(J))<>32THEN400
410 POKEBP(J),87:POKEBP(J)+FND(1),1:NEXT:NN=0
415 FORJ=1TO25
416 BC(J)=FNP(1):IFPEEK(BC(J))<>32THEN416
417 IFPEEK(BC(J)+1)=87THEN416
419 POKEBC(J),86:POKEBC(J)+FND(1),5:NEXT
420 GETR$:IFR$<>" "THEN420
430 DR=0:TX=TI
440 :
445 DR=2
450 POKE37154,127:R=NOTPEEK(37151)AND60-((PEEK(37152)AND128)=0)
453 IFPEEK(37152)=119THENR=1
454 POKE37154,255:R=R*2
455 DR=INT(R):IFR<>0THENDR=INT(LOG(DR)/LOG(2)):IFDR=5THENDR=2
460 IFPEEK(37152)=119THENDR=2
470 IFR=0THEN600
475 IFDR>4THENDR=0
480 :
500 POKEFNS(1),32:PX=PX+XM(DR):PY=PY+YM(DR)

```

```

510 IFPX<0THENPX=21
520 IFPX>21THENPX=0
530 IFPY<5THENPY=22
540 IFPY>22THENPY=5
550 X=FNN(1)
560 IFX=32THENPOKEFNS(1),90:POKE
(FNS(1)+FND(1)),1:GOTO600
570 IFX=42ORX=86THEN960
580 GOTO890
590 :
600 NN=NN+1:IFNN>INT(NB)THENNN=1

610 IFBS(NN)=0THEN600
620 TG=INT(TI*1.03)-TX
630 IFTG>B4(NN)THENNN=NN:GOTO720

640 IFBS(NN)>2THEN690
650 IFTG>BT(NN)THENBS(NN)=2
660 IFTG>B3(NN)THENBS(NN)=3
670 IFBS(NN)=1THEN450
680 IFBS(NN)=2THENPOKEBP(NN),215
:POKEBP(NN)+FND(1),1:GOTO450

690 IFBS(NN)=3THENPOKEBP(NN),87:
POKEBP(NN)+FND(1),1:BS(NN)=4
:GOTO450
700 IFBS(NN)=4THENPOKEBP(NN),215
:POKEBP(NN)+FND(1),1:BS(NN)=
3:GOTO450
710 :
720 TQ=TI:PD=0
725 X$="{OFF}" + LEFT$(S$,FNY(BP(N
1)) - 2) + LEFT$(Q$,FNX(BP(N1)) -
1)
730 BS(N1)=0:N2=0:PRINTX$;XP$;
740 FORJ=1TONB:X=PEEK(BP(J)):IFB
S(J)=0THEN760
750 IFX<>87ANDX<>215ANDX<>218THE
NN2=J
760 NEXT:IFFNN(1)<>90ANDFNN(1)<>
218THENPD=1
770 PRINTX$;XR$;
780 FORJ=1TONB:IFPEEK(BP(J))=32A
NDBS(J)<>0THENPOKEBP(J),87-1
28*(BS(J)>1)
790 NEXT:BN=BN-1
800 IFPD=1THEN960
810 IFBN=0THEN840
820 IFN2=0THENTX=TX+(TI-TQ):GOTO
450
830 N1=N2:GOTO725
840 PRINT"{HOME}{ 3 GIU' }{RVS}";
TAB(29);
850 FORJ=1TO11:PRINT"{ 2 DES}
{RVS}COMPLETATO{ 12 SIN}";:F
ORK=1TO100:NEXT
860 PRINT"{RVS}{ 12 SPAZI}
{ 12 SIN}";:FORK=1TO100:NEXT
:NEXT

```

```

870 SC=SC-10*(INT(NB)-BG):IFSC<0
THENSC=0
880 PRINT"{HOME}{RVS}";TAB(15);S
C;:GOTO280
885 :
890 BG=BG+1:TQ=TI:POKEFNS(1),218
:POKE(FNS(1)+FND(1)),1
895 FORJ=1TONB:IFPEEK(BP(J))=218
THENAJ=BS(J):BS(J)=0
900 NEXT
910 IFAJ=4THENAJ=3
920 SC=SC+10*AJ:PRINT"{HOME}
{RVS}";TAB(15);SC
930 BN=BN-1:IFBN=0THEN840
940 TX=TX+(TI-TQ):GOTO450
950 :
960 TQ=TI:FORJ=1TO20:POKEFNS(1),
42:FORK=1TO25:NEXT:POKEFNS(1
),170
970 FORK=1TO25:NEXT:NEXT:POKEFNS
(1),32:NL=NL-1
980 POKEFNV(1)+75+NL,160:DR=0:PX
=11:PY=11
990 IFNL=0THEN1045
1000 IFBN=0THEN840
1010 GETR$:IFR$<>" "THEN1010
1020 FORJ=1TONB:IFPEEK(BP(J))=32
ANDBS(J)<>0THENPOKEBP(J),87
-128*(BS(J)>1)
1030 NEXT
1040 POKEFNS(1),90:TX=TX+(TI-TQ)
:GOTO450
1045 IFSC>HSTHENHS=SC:PRINT"
{HOME}{GIU' }{RVS}";TAB(16);
HS;
1050 FORJ=1TO1500:NEXT:PRINT"
{HOME}{WHT}{ 4 GIU' }{RVS}";
"FINE GIOCO{GIU' }{WHT}
{ 10 SIN}ANCORA? ";
1060 PRINT"S/N{ 3 SIN}";
1080 PRINT"{RVS}S/{OFF}N
{ 3 SIN}";
1081 FORJ=1TO99:NEXT
1082 PRINT"{OFF}S{RVS}/N
{ 3 SIN}";
1083 FORJ=1TO99:NEXT
1084 GETR$:IFR$="S"THEN1110
1090 IFR$<>"N"THEN1080
1100 PRINT"{CLR}{WHT}{ 2 SPAZI}
CIAO!!{ 2 SPAZI}":END
1110 GOSUB1410:GOTO150
1400 GETR$:IFR$<>CHR$(13)THEN140
0
1410 PRINTCHR$(14);"{OFF}{CLR}
{WHT}LIVELLO DI DIFFICOLTA'
{GIU' }(0-3)"
1420 GETR$:IFR$<"0"ORR$>"3"THEN1
420
1430 NL=4-VAL(R$):RETURN

```

# Il pescatore

Ondeggiare prigramente sulle onde aspettando che un pesce abocchi è un ottimo mezzo di relax. In questo gioco tuttavia il sogno di un pescatore si trasforma in incubo quando gli squali lo prendono per esca e nubi temporalesche lo minacciano con enormi gocce.

È utile quindi portare un ombrello ed un robusto remo per tenere a bada gli squali.

Una tranquilla giornata di pesca diventa una lotta contro le avverse forze della natura impersonate da un cielo minaccioso e da voraci squali. Scopo del gioco è sopravvivere. Si accumulano punti colpendo gli squali col remo e fermando le gocce con l'ombrello.

Si comincia con tre pescatori ogni volta che uno squalo o una goccia raggiungono il pescatore questo è irrimediabilmente perso, tuttavia ogni 2000 punti si guadagna una vita.

Per il controllo dei movimenti si usano tre tasti. Per andare a sinistra e a destra si utilizzano i tasti (e). La barra assolve 2 funzioni: quella di muovere il remo e quella di aprire l'ombrello. A seconda della necessità la barra aziona o l'ombrello o il remo. Se si sta usando il remo e si tiene premuta la barra questo continua a menar mazzate. Attenzione all'uso dell'ombrello perché tenendo

premuta la barra questo non rimane aperto in continuazione mai si apre e si chiude tre volte sole durante la caduta di una goccia. Quando si usa l'ombrello bisognerà quindi aprirlo solo al momento opportuno.

Se il gioco fosse troppo veloce o troppo lento si potrà benissimo rallentare od accelerare l'azione modificando il valore della variabile DE nella linea 30 del 2° programma. Se si volessero invece avere più pescatori a disposizione basterà modificare il valore della variabile GL nella linea 100. Si noti che le vite a disposizione sono visualizzate sotto forma di faccine nell'angolo in alto a sinistra dello schermo, proprio sotto il punteggio.

Come inserire il programma

Il programma gira su VIC 20 non espanso ed è diviso in due parti.

La prima parte ridefinisce il set dei caratteri mentre la seconda costituisce il programma vero e proprio.

Una volta digitato correttamente il 1° programma lo si salvi su cassetta e si passi a digitare la seconda parte. Il secondo programma andrà salvato su cassetta subito dopo il primo. Per caricare il gioco basterà caricare la prima parte e dare il RUN e provvederà poi questa a caricare automaticamente la seconda parte e dare il RUN.

## Programma I

```

1 PRINT"{CLR}{ 3 GIU' }
  { 4 SPAZI}IL PESCATORE"
2 PRINT"{ 4 GIU' }CARICAMENTO CAR
  ATTERI"
10 FORT=7168TO7168+62*8-1:READA:
  POKET,A:NEXT
20 DATA14,62,254,62,14,2,6,6
30 DATA14,14,14,22,22,22,54,54
40 DATA118,118,246,254,246,246,2
  46,246
50 DATA0,0,0,1,1,3,7,7
60 DATA246,246,246,254,246,246,2
  46,246
70 DATA15,31,63,127,255,255,255,
  255

```

```

80 DATA0,0,0,0,0,3,7,31
90 DATA246,246,6,254,254,6,6,15
100 DATA255,255,8,255,255,0,0,0
110 DATA127,127,64,255,255,0,0,0
120 DATA255,255,127,127,63,31,15
  ,7
130 DATA170,255,85,0,255,85,0,17
  0
140 DATA255,255,254,254,252,248,
  240,224
150 DATA6,12,24,240,192,0,0,0
160 DATA0,0,0,192,240,24,12,6
170 DATA96,48,24,15,3,0,0,0
180 DATA0,0,0,3,15,24,48,96

```

```

190 DATA40,40,170,60,28,252,12,6
0
200 DATA12,40,40,248,248,40,20,4
0
210 DATA40,40,170,60,52,63,48,60
220 DATA48,40,40,47,47,40,20,40
230 DATA32,48,190,187,252,255,19
0,128
240 DATA4,12,125,221,63,255,125,
1
250 DATA0,0,0,0,0,0,7,195
260 DATA0,0,0,0,1,13,109,255
270 DATA0,0,0,0,128,224,248,248
280 DATA0,0,0,0,1,15,15,15
290 DATA7,3,15,63,255,255,255,25
5
300 DATA199,255,255,255,255,255,
255,255
310 DATA255,255,255,255,255,255,
255,255
320 DATA0,192,240,240,224,248,25
2,248
330 DATA255,7,31,127,3,24,31,31
340 DATA0,0,0,0,0,0,0,0
350 DATA255,255,255,255,255,127,
255,255,0,0,0,0,0,0,0
360 DATA255,255,255,255,252,255,
255,224
370 DATA255,192,240,252,0,0,128,
0
380 DATA0,0,63,0,0,0,0,0
390 DATA3,3,193,1,1,1,0,0
400 DATA0,0,0,0,0,3,12,0
410 DATA0,0,0,48,192,0,0,0
420 DATA127,127,63,63,31,15,7,3

```

```

430 DATA1,2,4,8,16,32,64,128
440 DATA0,0,0,0,4,4,8,8
450 DATA255,127,31,7,0,0,0,4
460 DATA16,16,32,32,0,0,0,0
470 DATA4,4,4,4,4,4,4,0
475 DATA255,199,189,207,245,143,
255,255
480 DATA255,129,145,169,169,169,
145,255
490 DATA255,129,145,177,145,145,
185,255
500 DATA255,129,153,165,137,145,
189,255
510 DATA255,129,185,137,153,137,
185,255
520 DATA255,129,153,169,189,137,
137,255
530 DATA255,129,189,161,185,133,
185,255
540 DATA255,129,157,161,185,165,
153,255
550 DATA255,129,189,133,137,145,
145,255
560 DATA255,129,153,165,153,165,
153,255
570 DATA255,129,153,165,157,133,
185,255
580 DATA20,58,28,119,8,54,8,54
590 DATA8,8,28,20,58,62,62,28
600 DATA28,62,127,73,8,8,40,16
610 DATA16,2,32,136,80,42,116,56
620 REM
630 POKE52,PEEK(52)-2:POKE56,PEE
K(56)-2
700 POKE198,1:POKE631,131:END

```

## Programma 2

```

10 CL$="{WHT}WXY{ 5 SIN}{GIU'}Z[
EE]↑{ 5 SIN}{GIU'}+!# $"
20 DEF FNRN(X)=INT(RND(1)*X)
30 V=36878:NO=V-1:S=V-2:S2=V-3:S
3=V-4:CO=30720:EG=2000:TT=22:
T6=256:Z=32:DE=60
40 GOTO7000
100 GL=3:SC=.
105 EG=2000
110 POKE36869,255
120 PRINT"{CLR}{ 11 DES}
{ 18 GIU'}{GRN}@{SIN}{GIU'}
{WHT}A{ 2 SIN}{GIU'}CB
{ 3 SIN}{GIU'}FED{ 3 SIN}
{GIU'}IHG{ 4 SIN}{GIU'}{RED}
JKKKL";
130 POKE646,10:PRINT"{ 4 SIN}KKK
":PRINT"{CYN}]]]]]]]]]]]]]]]]]]
]]]]]]{SIN}{INST}]{HOME}
{ 3 GIU'}":POKEV,15+16*9
140 POKE36879,238

```

```

150 PRINT"{HOME}{ 6 GIU'}
{ 2 DES}"CL$"{ 7 SU}
{ 4 DES}"CL$"{ 3 GIU'}
{ 3 DES}"CL$;
160 PRINT"{YEL}{ 8 SU}%&]]
{ 4 SIN}{GIU'}'()]{ 3 SIN}
{GIU'}*+,{ 2 SIN}{GIU'}-."
170 PRINT"{HOME}{BLK}/"SC;:POKE6
46,8:PRINT"{HOME}{GIU'}";:IF
GL>1THENFORT=1TOGL-1:PRINT"S
";:NEXT
180 LO=8128:POKELO+CO,10:POKELO,
18:POKELO-TT,17:POKELO-TT+CO
,10:POKELO-1,15
190 POKELO-1+CO,.:CL=LO-1:POKE81
32+CO,10:POKE8110+CO,10:POKE
8133+CO,.
200 TY=FNRN(2)+1:ONTYGOTO210,300
210 X=FNRN(2)+1:ONXGOTO220,230

```

```

220 BC=8142:EC=8149:SP=1:DD=21:G
OTO240
230 BC=8163:EC=8155:SP=-1:DD=22
240 FORDL=BCTOECSTEPSP:POKEDL,DD
:POKEDL+CO,.
250 GOSUB1000:POKEDL,Z:NEXT:IFSD
<>196THEN3000
260 SD=.:GOTO200
300 NU=.:Y=59:C=6:X=FNRN(2)+1:ON
XGOTO310,320
310 B=7776:E=8084:GOTO330
320 B=7758:E=8088
330 FORDL=BTOESTEPTT:POKEDL+CO,C
:POKEDL,Y:GOSUB1000
340 POKEDL,Z:NEXT:IFSP<>T6THEN30
00
350 SP=.:GOTO200
1000 CK=PEEK(197):IFCK=64THENFOR
R=0TODE:NEXT:RETURN
1010 IFCK=29ANDLO=8132THEN1500
1020 IFCK=37ANDLO=8128THEN1750
1030 IFCK=ZTHENONTYGOTO2000,3500

1040 RETURN
1500 POKELO,Z:POKELO-TT,Z:POKECL
,Z:LO=8128:CL=LO-1
1510 POKELO,18:POKELO-TT,17:POKE
CL,15:FORS=130TO150STEP2:P
OKES2,SD:NEXT:POKES2,.
1520 RETURN
1750 POKELO,Z:POKELO-TT,Z:POKECL
,Z:LO=8132:CL=LO+1
1760 POKELO,20:POKELO-TT,19:POKE
CL,13
1770 FORSD=150TO130STEP-2:POKES2
,SD:NEXT:POKES2,.:RETURN
2000 POKECL,PEEK(CL)+1:FORS=250
TO200STEP-10:POKENO,SD:NEXT
:IFPEEK(CL+TT)=DDTHEN2100
2010 POKECL,PEEK(CL)-1:POKENO,.:
RETURN
2100 SC=SC+75:PRINT"{HOME}{BLK}/
"SC:GOSUB4000
2110 POKEDL,58:FORS=254TO198STE
P-2:POKEDL+CO,FNRN(8):POKEN
O,SD:NEXT
2120 POKENO,.:GOTO2010
3000 GL=GL-1:FORT=130TO254STEP2:
POKES,T:POKENO,T:POKEV,15+F
NRN(16)*16:NEXT
3010 POKELO-TT,218:FORT=15TO0STE
P-.2:POKES,.:POKENO,160:POK
EV,T+FNRN(16)*16:NEXT
3020 IFGL=.THEN7000
3030 POKENO,.:GOTO120
3500 IFNU>2THENRETURN
3510 NU=NU+1:POKELO-44,60:POKELO
-44+CO,4:FORS=150TO180STEP
10:POKES2,SD:NEXT:POKES2,.
3520 IFPEEK(LO-66)=59THEN3600
3530 POKE(LO-44),Z:RETURN
3600 SC=SC+50:PRINT"{HOME}{BLK}/
"SC:GOSUB4000:POKELO-66,61
3610 FORSP=200TO254STEP2:POKES,S
P:NEXT:POKES,.
3620 POKELO-66,Z:RETURN
4000 IFSC>EGTHENGL=GL+1:DE=DE-4
:EG=EG+2000:POKE7700+GL,19:
POKE7700+CO+GL,8:GOTO4020
4010 RETURN
4020 FORT=130TO230STEP10:FORR=T+
10TOTSTEP-1:POKES,T:NEXTR,T
:POKES,.:RETURN
7000 POKE36869,240:PRINTCHR$(8):
IFSC>HSTHENHS=SC
7010 POKE36879,8:POKE646,10:PRIN
T"{CLR}{ 3 SPAZI}{<A>
{ 2 *}{<S>":PRINT"
{ 3 SPAZI}_-<A>*<X>"
7020 PRINT"{ 3 SPAZI}_-<Z>[<R>]
*-<R>*<R>{ 2 *}{<R>]
{ 2 *}{< 2 R>]*-<S>]
{ 6 SPAZI}_-<A>[<W>][<A>]
[<W>][<D>]-[<A>][<S>][<Q>]
[<S>][<A>][<W>]-[<A>][<X>]
{ 6 SPAZI}{ 5 _}{SPAZI}
{ 8 _}{<Z>][<S>]"
7030 PRINT"<A>{ 2 *}+<X>]
[<Q>][<W>][<Z>][< 2 E>]
[<X>][<Z>][<X>][<Z>][<X>]
[<Z>][<E>]*-<X>":PRINT"-
[<A>]*-<X>]{ 2 _}"
7040 PRINT"-[<Z>][< 2 R>]*-<W>]
[<Z>]*-<R>]*-<R>]*-<R>]
{ 2 *}{<R>]*-<R>]{ 2 *}
[<S>]-[<A>][<W>]-*-<W>]
[<A>][<S>]-[<V>]-[<A>][<W>]
{ 2 SPAZI}_-<D>]-[<A>][<S>]
_"
7050 PRINT"{ 3 _}{<Q>]*{ 4 _}*
[<W>]{ 5 _}{ 4 _}{<Z>]
[<X>][<Z>][<E>]*-<E>][<X>]
[<Z>][<E>]*-<E>][<X>][<Z>]
[< 5 E>][<X>][<Z>][<X>]"
7055 PRINT"{WHT}{ 5 DES}IL PESCA
TORE"
7060 PRINT"{RED}{ 7 DES}PUNTI:"S
C:PRINT"{GIU'}{GRN}
{ 6 DES}RECORD:"HS
7070 PRINT"{PUR}{GIU'}{ 3 DES}PR
EMI UN TASTO"
7080 PRINT"{RVS}{WHT}{ 7 SPAZI}
COMANDI{ 8 SPAZI}{OFF}{PUR}
{ 5 SPAZI}<-SINISTRA":PRINT
"{GRN}{ 5 SPAZI}>-DESTRA"
7090 PRINT"{WHT}BARRA= REMO O OM
BRELLO{SU}":POKENO,.
7100 POKE36878,(FNRN(14)+2)*16:I
FPEEK(197)=64THEN7100
7110 GOTO100

```

# Allegre, Fresche, Spiritose, Pratiche.

HOME COMPUTER  
**hce**  
LA RIVISTA DEL COMPUTER IN CASA  
SETTEMBRE 1984 L. 3.500



MUSICA, MUSICA, MUSICA: CHI  
CON CHE COMPUTER, CON QU  
ADAM, UN HOMENATO PE  
ARRIVANO I NUOVI CO

VIC 20, SEGA, SPECTRUM, TI 99/4A, CMB



**VIDEO GIOCHI** 18  
LA PRIMA E UNICA RIVISTA ITALIANA DI VIDEOGAMES E GIOCOMPUTER  
SETTEMBRE 1984 L. 3.500

**SUPER CROSS FORCE**  
**HIT BIT: I VOSTRI COMPUTER RECORD**

**TUTTO CHICAGO**

**INTERVISTA A DON BLUTH  
IL CREATORE DI DRAGON'S LAIR**

**IL NUOVO ADAM**



Con tutta la competenza del  
**GRUPPO EDITORIALE  
JACKSON**